

Язык Клаус

справочное руководство

Памяти Никлауса Вирта.

*С благодарностью Дмитрию Тарасевичу и
Анне Михеевой.*

Алфавит и кодировка

Язык Клаус поддерживает только кодировку UTF-8. Эта кодировка используется для исходных файлов, для хранения строковых данных в оперативной памяти, для работы функций файлового ввода-вывода стандартной библиотеки, а также во всех остальных случаях.

Алфавит языка включает в себя:

- пробельные символы: пробел, табуляция, возврат каретки, перевод строки;
- буквы латиницы: **A-Z** и **a-z**;
- буквы кириллицы: **А-Я** и **а-я**;
- арабские цифры: **0-9**;
- символы: `_ + - * \ / % ^ = < > ! & | ~ : ; . , () [] { } $ # ` ' " №`

Строковые и символьные литералы в исходном коде могут содержать любые символы, допустимые в кодировке UTF-8, за исключением символов возврата каретки, перевода строки и нулевого символа.

Язык Клаус не регистрочувствительный — т. е., прописные буквы в ключевых словах и идентификаторах считаются эквивалентными строчным.

Лексика

Лексемы языка Клаус всегда можно отделять друг от друга пробельными символами — пробелами, табуляциями, символами перевода строки и возврата каретки. Перевод строки и возврат каретки равнозначны пробелу и не имеют специального синтаксического значения, с той оговоркой, что:

- перевод строки и возврат каретки недопустимы внутри строковых и символьных литералов;
- перевод строки и возврат каретки считаются окончанием однострочного комментария.

В остальном, разбиение исходного текста на строки остаётся полностью на усмотрение программиста.

В тех случаях, когда Клаус может распознать лексемы однозначно, отделять их друг от друга пробельными символами не требуется.

В языке Клаус есть следующие виды лексических единиц:

- **Ключевое слово** (зарезервированное слово) — слово, состоящее только из букв кириллицы и имеющее специальное значение в синтаксисе языка. Некоторые ключевые слова могут иметь несколько разных форм-синонимов, имеющих одинаковое значение, например, **любой** **любая** **любое** или **целое** **целых**, и т. п. Перечень ключевых слов и их синонимов приведён в соответствующем разделе.

- **Идентификатор** — слово, состоящее из букв латиницы или кириллицы, арабских цифр и символов нижнего подчёркивания; первым символом идентификатора не может быть цифра. Идентификаторы используются в качестве имён программ и их параметров, подпрограмм и их параметров, типов данных, констант, переменных, исключений и их параметров, полей структур.
- **Знак языка** — один не буквенный и не цифровой символ, либо комбинация из двух таких символов. К знакам языка относятся знаки арифметических и логических операций, круглые и квадратные скобки, точка, двоеточие, а также другие знаки, имеющие специальное синтаксическое значение.
- **Литерал** — буквальное значение того или иного типа данных в исходном коде программы. Форматы литералов описаны для каждого из типов данных в соответствующем разделе.
- **Однострочный комментарий** — любой текст, начиная от удвоенной косой черты «//» и до конца строки — т. е., до ближайшего символа перевода строки или возврата каретки.
- **Многострочный комментарий** — любой текст, начиная от открывающей фигурной скобки «{» и до ближайшей закрывающей фигурной скобки «}». Вложенные многострочные комментарии не допускаются.

Ключевые слова языка

Ключевые слова, перечисленные в одной строке, являются синонимами — т. е., в исходном тексте программист может использовать любую из перечисленных форм по своему усмотрению. Это позволяет заметно, хотя и не полностью, приблизить текст программы к нормам русского языка.

Пример:

```

типы
с: строка;
ц: целое;
тПервый = массив строк;
тВторой = словарь целых ключ символ;
тТретий = массив словарей массивов структур
    поле1: дробное;
    поле2: логическое;
окончание ключ строка;

```

Использование тех или иных форм ключевых слов остаётся на усмотрение программиста: конструкции «словарей массив строка» и «словарь массивов строк» с точки зрения синтаксиса полностью эквивалентны. Авторы, однако, надеются, что из соображений читаемости кода большинство программистов всегда предпочтут второй вариант.

Перечень ключевых слов

```

программа
начало
окончание конец
переменная переменные
константа константы
тип типы
символ символы символов
строка строки строк
целое целые целых
дробное дробные дробных
момент моменты моментов
логическое логические логических
объект объекты объектов

```

да
нет
массив массивы массивов
словарь словари словарей
структура структуры структур
ключ
процедура
функция
ничего
прервать
продолжить
вернуть вернуться
если
то
иначе
для
от
до
цикл
обратный
пока
выбор
из
исключение исключения
ошибка
бросить
сообщение
завершить
когда
тогда
напоследок
любой любая любое
вх
вых
вв
каждый каждого каждой
есть
нету
пусто
точность
дальше

Структура исходного кода

Программа на языке Клаус, а также любая подпрограмма — процедура или функция — состоит из заголовка, раздела определений и составного блока инструкций, представляющего собой тело программы или подпрограммы.

Определение программы

Заголовок программы начинается ключевым словом **программа**, за которым следует идентификатор — имя программы. Далее, если программа принимает аргументы командной строки, в круглых скобках следует определение единственного параметра, принимаемого программой, который всегда должен быть объявлен как входной и иметь тип **массив строк**.

Заголовок программы завершается знаком «;».

```

программа Пустая;
начало
    ничего;
окончание.

программа Эхо (арг: массив строк);
переменные
    к: целое;
    рзд: строка = "";
начало
    для к от 0 до длина (арг) - 1 цикл начало
        вывести (рзд, арг[к]);
        рзд := " ";
    конец;
    вывести (НС);
окончание.

```

Между заголовком и телом программы могут располагаться определения переменных, констант, типов данных и подпрограмм. При этом каждое из определений может обращаться к именам, объявленным выше него по тексту исходного кода.

Тело программы представляет собой составной блок инструкций, который должен содержать одну или более инструкций Клаус, а также может содержать секции обработки исключений.

После окончания тела программы должен следовать знак «. ».

Определение подпрограммы

В языке Клаус существует две разновидности подпрограмм: процедуры и функции. Они полностью идентичны во всём, за тем исключением, что функция возвращает значение, а процедура не возвращает.

Подпрограммы могут быть определены в разделе определений основной программы, либо в разделах определений других подпрограмм с произвольным уровнем вложенности.

Заголовок процедуры начинается ключевым словом **процедура**, заголовок функции — словом **функция**.

Далее следует имя подпрограммы и, если подпрограмма принимает параметры, перечень определений параметров в круглых скобках. Для подпрограмм, не принимающих параметры, обязательно должны быть указаны пустые круглые скобки.

Затем, только для функций, должен следовать знак «:» и имя типа данных, возвращаемого функцией — ключевое слово или идентификатор. Безымянное определение типа данных в этом случае не допускается.

Заголовок подпрограммы завершается знаком «;».

Раздел определений подпрограммы и её тело устроены так же, как у основной программы. После окончания тела подпрограммы должен следовать знак «;».

Параметры подпрограмм

В Клаусе определено три режима передачи параметров в подпрограммы: входные, выходные и проходные параметры.

Значение входного параметра непосредственно передаётся в подпрограмму, для чего в подпрограмме создаётся локальная переменная. Подпрограмма может произвольным образом менять значение входного параметра, но эти изменения не выходят за пределы подпрограммы. Даже если в качестве входного параметра была указана переменная, значение этой переменной после выхода из подпрограммы всегда останется таким же, каким было перед вызовом.

В качестве входных параметров можно указывать переменные, константы, буквальные значения и выражения, включая вызовы функций.

Выходным или проходным параметром может служить только переменная или часть переменной — например, поле структуры или элемент массива. В этом случае в подпрограмму передаётся не значение, но ссылка на переменную или её часть. Если подпрограмма вносит изменение в значение такого параметра, то это новое значение будет присвоено той переменной из внешней области видимости, что была передана в подпрограмму в качестве параметра — т. е., такое присваивание будет иметь эффект после выхода из подпрограммы.

Различие между выходными и проходными параметрами состоит в том, что значения проходных параметров передаются в подпрограмму без изменений, а значения выходных обнуляются перед вызовом подпрограммы.

Определение параметра начинается ключевым словом **вх**, **вых** или **вв** для входных, выходных и проходных параметров, соответственно. Ключевое слово **вх** можно опустить, в таком случае параметр будет считаться входным.

Далее следует имя параметра или несколько имён через запятую, а затем знак «:» и имя типа данных. Если указано несколько имён параметров, то все они будут иметь одинаковый режим и тип. Для объявления параметров разных режимов или типов их определения нужно разделить знаком «;».

В качестве типов параметров можно использовать ключевые слова, соответствующие простым типам Клаус, а также идентификаторы ранее объявленных типов данных. Безымянные определения типов в данном случае не допускаются. Единственным исключением из этого правила является определение входного параметра основной программы, который всегда должен быть объявлен как **массив строк**.

Предварительное определение

Предварительное определение подпрограммы представляет собой её заголовок, после которого вместо тела подпрограммы указано ключевое слово **далее** и знак «;».

```
функция мояфункция(а, б: целое): дробное; далее;
```

Если в исходном коде дано такое предварительное определение, то в той же области видимости ниже по тексту должно располагаться полное определение этой подпрограммы, включая её тело, причём заголовок должен точно совпадать с заголовком в предварительном определении.

Предварительное определение используется в тех случаях, когда несколько подпрограмм должны вызывать друг друга рекурсивно.

Примеры

```
процедура пустая();
начало
  ничего;
окончание;
```

```
функция синус(х: дробное): дробное;
начало
  вернуть sin(х);
окончание;
```

```
функция корень(вх х: дробное; вых рез: дробное): логическое;
начало
  если х < 0 то вернуть нет;
  рез := х^(1/2);
  вернуть да;
окончание;
```

```
тип тТочка = структура
  х, у: целое;
конец;
```

```

функция точка(x, y: целое): тТочка;
переменная
    т: тТочка;
начало
    т.x := x;
    т.y := y;
    вернуть т;
окончание;

процедура подвинуть(вв т: тТочка; вх dx, ду: целое);
начало
    т.x += dx;
    т.y += ду;
окончание;
процедура первая(); дальше;

процедура вторая();
начало
    первая();
окончание;

процедура первая();
начало
    вторая(); // здесь будет бесконечная рекурсия
окончание;

```

Области видимости

Каждой подпрограмме, а также основной программе, соответствует своя область видимости идентификаторов. В пределах одной области видимости имена должны быть уникальны — т. е., ни одно имя не может совпадать с другим именем, объявленным в той же области видимости. Регистр букв в именах не учитывается — т. е., слова **имя**, **ИМЯ** и **Имя** считаются совпадающими.

Область видимости подпрограммы является вложенной по отношению к основной программе или к объемлющей подпрограмме, в которой она определена. Имена, объявленные во вложенной области видимости, могут совпадать с именами, объявленными в объемлющих областях видимости, но не друг с другом.

К области видимости основной программы относятся:

- Имя программы;
- Имя входного параметра программы, если он объявлен;
- Все имена, объявленные в области определений программы.

По аналогии, к области видимости подпрограммы относятся имя подпрограммы, имена её параметров и все локально объявленные имена.

Таким образом, имя подпрограммы принадлежит сразу двум областям видимости — той, в которой подпрограмма объявлена, и своей собственной. Следовательно, имя подпрограммы должно быть уникально в обеих этих областях.

При обращении к идентификатору в исходном коде программы имя разрешается по следующим правилам:

- Выполняется поиск имени в локальной области видимости, среди тех имён, что объявлены выше по тексту исходного кода;
- Если имя не найдено, то выполняется поиск выше по тексту во всех объемлющих областях видимости, по порядку изнутри наружу;

Иными словами, в каждой области видимости видны её собственные имена, а также все имена из объемлющих областей видимости, если они не были перекрыты собственными. Если во вложенной области видимости объявлено имя, совпадающее с именем из внешней области видимости, то обращение к последнему будет невозможно, т. к. локальный идентификатор скрывает за собой внешний.

Имена, объявленные во вложенных областях видимости, никогда не видны из объемлющих областей.

В настоящей версии языка определена единственная область видимости, внешняя по отношению к основной программе, — область видимости с именем **Клаус**, в которой объявлены типы, константы, переменные, исключения и подпрограммы стандартной библиотеки языка.

Определения

Глобальные определения типов данных, констант и переменных размещаются разделе определений основной программы — между её заголовком и началом составного блока инструкций. Такие определения принадлежат глобальной области видимости, т. е., они видны в теле основной программы, а также во всех вложенных областях видимости — в процедурах и функциях, — если не перекрыты в них локальными определениями.

Любая подпрограмма имеет собственный раздел определений между её заголовком и телом, в котором также могут быть определены типы, константы и переменные. Эти определения являются локальными по отношению к подпрограмме — т. е., они видны только в ней, а также во вложенных в неё подпрограммах, но не видны из объемлющих областей видимости.

Во всех случаях видны только те определения, которые были даны выше по тексту исходного кода. Глобальная переменная или константа, объявленная после подпрограммы, не будет видна внутри этой подпрограммы.

```
программа Пример;  
тип мойТип = массив целых;  
константа Пи = 3.1415926;  
переменная мт: мойТип;  
начало  
    ничего;  
окончание.
```

Определения типов

Секция определений типов должна располагаться в разделе определений программы или подпрограммы. Секция начинается ключевым словом **тип** или **типы**, за которым следует одно или несколько объявлений типов. Таких секций в разделе определений может быть сколько угодно, они могут быть расположены подряд или вперемежку с другими секциями и подпрограммами.

Объявление типа начинается с идентификатора (или нескольких идентификаторов, разделённых знаком «/»), за которыми следует знак «=», а за ним — определение типа. Определением может служить ключевое слово, обозначающее простой тип Клаус, идентификатор ранее объявленного типа, либо определение составного типа — массива, словаря или структуры. После каждого определения должен следовать знак «;».

```
программа Пример;  
типы  
    ЦЧ = целое;  
    ЦЧ2 = ЦЧ;  
    МЦЧ = массив целых;  
    МСтр = массив структур  
        п1, п2: целое;  
        п3: строка;  
окончание;
```

```
начало
ничего;
окончание.
```

Такие определения типов можно использовать при объявлении переменных, элементов массивов и словарей, полей структур, параметров подпрограмм и возвращаемых значений функций. Параметры исключений могут иметь только простой тип, обозначаемый ключевым словом языка.

Объявление типов, имеющих различные имена, но эквивалентные определения, зачастую используется для улучшения читаемости исходного кода программы.

Если в объявлении типа данных указано несколько идентификаторов через «/», то все эти идентификаторы являются синонимами, т.е. в исходном коде для улучшения читаемости может использоваться любой из них, на усмотрение программиста.

```
программа Пример;
тип
    Точка/Точек = структура
        x: дробное;
        y: дробное;
    окончание;
переменные
    т: Точка;
    мт: массив Точек;
начало
ничего;
окончание.
```

Синтаксис определений составных типов подробно описан в разделе **Типы данных**.

Определения констант

Константа представляет собой именованное значение простого типа, которое определяется при компиляции и не может меняться в процессе выполнения программы. Константы не имеют другого смысла, кроме улучшения читаемости исходного кода программы.

Константы составных типов недопустимы в нынешней версии языка.

Секция определений констант должна располагаться в разделе определений программы или подпрограммы. Секция начинается ключевым словом **константа** или **константы**, за которым следует одно или несколько объявлений констант. Таких секций в разделе определений может быть сколько угодно, они могут быть расположены подряд или вперемежку с другими секциями и подпрограммами.

Объявление константы начинается с идентификатора, за которым следует знак «=», а за ним — буквальное значение, имя ранее определённой константы или выражение. После каждого определения должен следовать знак «;».

Тип константы не может быть указан явно — он определяется исходя из типа указанного буквального значения или выражения. Поэтому, например, если требуется объявить константу — дробное число со значением, равным единице, то буквальное значение должно быть специально указано как дробное: **1.0** или **1÷0**.

Если определением константы служит выражение, то это выражение должно быть вычислимо на этапе компиляции — т.е., оно не может содержать иных операндов, кроме буквальных значений и имён ранее объявленных констант. Обращения к переменным и вызовы функций в таком выражении недопустимы.

```
программа Пример;
константы
    целыйНуль = 0;
```

```
дробныйНуль = 0.0;
неЧисло = 0 / 0;
бесконечность = 1 / 0;
Пи = 3.1415926;
прямойУгол = Пи / 2;
начало
    ничего;
окончание.
```

Определения переменных

Переменная представляет собой именованное значение простого или составного типа, которое может меняться в процессе выполнения программы. Начальное значение переменной может быть указано при объявлении. Если оно не указано, то при входе в область видимости переменная получает нулевое начальное значение, как это описано в разделе **Типы данных**.

Для переменных составных типов начальное значение не может быть указано.

Секция определений переменных должна располагаться в разделе определений программы или подпрограммы. Секция начинается ключевым словом **переменная** или **переменные**, за которым следует одно или несколько объявлений переменных. Таких секций в разделе определений может быть сколько угодно, они могут быть расположены подряд или вперемежку с другими секциями и подпрограммами.

Объявление переменной начинается с идентификатора или нескольких идентификаторов, перечисленных через запятую — так можно объявить несколько однотипных переменных. Затем следует знак «:», а за ним — определение типа переменной. После этого, если переменная имеет простой тип, может быть указан знак «=» и начальное значение. Каждое объявление завершается знаком «;».

В качестве определения типа допускается одно из ключевых слов простых типов, идентификатор ранее объявленного типа, либо безымянное определение составного типа.

Начальное значение должно быть либо буквальным, либо выражением, вычислимым на этапе компиляции — т. е., в этом выражении можно использовать только буквальные значения и константы, а переменные и вызовы функций не допускаются. Если одновременно объявляются несколько переменных через запятую, то все они получат одинаковые начальные значения.

Особым случаем является объявление переменной в инструкции перехвата исключения **когда...тогда**. В этом случае переменная объявляется в ближайшей объемлющей области видимости — т. е., в разделе определений той подпрограммы, которая содержит инструкцию перехвата. Однако, значение такой переменной фактически создаётся только при входе в инструкцию перехвата — т. е., только если возникло соответствующее исключение — и уничтожается при выходе из неё. Это приводит к тому, что обращение к переменной вне тела инструкции перехвата не приводит к синтаксической ошибке при компиляции, но приводит к возникновению исключения **ОшибкаВыполнения** при выполнении программы. Возможно, в следующих версиях языка такое некорректное поведение будет исправлено.

```
программа Пример;
тип
    комплексное = структура
        д: дробное;
        м: дробное;
окончание;
переменные
    стр: строка = "Новая строка";
    н, м: целое = 1;
    к: комплексное;
    точки: словарь структур
        х, у: дробное;
окончание ключ строка;
```

```
начало
ничего;
окончание.
```

Определения исключений

Всплывающее исключение представляет собой специальный именованный объект, который создаётся для того, чтобы прервать последовательность выполнения программы в случае возникновения нештатной ситуации и задействовать средства обработки ошибок.

Секция определений исключений должна располагаться в разделе определений программы или подпрограммы. Секция начинается ключевым словом **исключение** или **исключения**, за которым следует одно или несколько объявлений исключений. Таких секций в разделе определений может быть сколько угодно, они могут быть расположены подряд или вперемежку с другими секциями и подпрограммами.

Объявление исключения начинается с идентификатора. Далее, если исключение принимает параметры, то их определения должны быть указаны в круглых скобках. Затем может следовать ключевое слово **сообщение**, а за ним текст сообщения об ошибке — буквальное значение или выражение строкового типа. Каждое объявление завершается знаком «;».

```
программа Пример;
исключения
    Ошибка1;
    Ошибка2 сообщение "Вторая ошибка";
    ОшибкаВТексте(стр, симв: целое; чтоНеТак: строка)
        сообщение "Ошибка в строке %(стр), символ %(симв). %(чтоНеТак)";
начало
    ошибка ОшибкаВТексте(5, 10, "Пропущена запятая.");
окончание.
```

Параметры исключений объявляются по тем же правилам, что и параметры подпрограмм, с двумя оговорками:

- они могут быть только простых типов;
- они могут быть только входными.

Если текст сообщения об ошибке не указан в определении исключения, его можно указать при создании исключения. Если он не указан и там, то в качестве сообщения об ошибке используется имя исключения.

Если исключение имеет параметры, то текст сообщения может содержать спецификаторы вида **%(имя)**, где имя — имя параметра исключения. При создании исключения такие спецификаторы будут заменены на строковые представления значений параметров.

Типы данных

Приведение типов

Значения простых типов можно преобразовывать в другие простые типы. В большинстве случаев такое преобразование должно быть выполнено явно. Для этого используется конструкция приведения типа, которая записывается как ключевое слово — имя типа данных, за которым в круглых скобках указывается исходное значение или выражение. Такая конструкция конвертирует исходное значение в указанный тип, а если это невозможно, создаётся исключение.

```

переменные
  д: дробное = 123.45;
  ц: целое = 10;
  с: строка = "yes";
  л: логическое;
начало
  л := логическое(с); // л = да
  с := строка(ц);     // с = "10"
  ц := целое(д);     // ц = 123
  д := ц;             // д = 123.0 (неявное преобразование)
  д := дробное("Ку"); // исключение ОшибкаКонвертации
окончание;

```

Преобразование типов данных выполняется по правилам, описанным ниже для каждого из простых типов.

Неявное преобразование допускается в двух случаях:

- при преобразовании символа в строку, но не наоборот;
- при преобразовании целого числа в дробное число, но не наоборот.

В нынешней версии языка для приведения типа не допускается использовать имена ранее объявленных типов данных — допустимы только ключевые слова простых типов. Кроме того, не допускается явное приведение составных типов, хотя в некоторых случаях возможно неявное преобразование.

Простые типы

Символ

Переменная символьного типа содержит кодпункт Юникод в формате 32-битного целого числа баз знака. Нулевые значения допускаются.

Синтаксис

```

переменные
  п1: символ;
  п2: массив символов;

```

Ключевые слова **символ** и **символов** полностью эквивалентны и взаимозаменяемы.

Формат литерала

```

'X' // символ в кодировке UTF-8, заключённый в одиночные апострофы
№2A // код символа в формате 16-ричного целого числа
#2A // код символа в формате 16-ричного целого числа

```

Начальное значение

Символ с нулевым кодом.

Преобразование типов

При приведении к целочисленному типу возвращает численное значение кода символа.

При приведении к строке возвращает строку, содержащую один символ в кодировке UTF-8, за исключением нулевого значения, которое приводится к пустой строке.

Строка

Переменная строкового типа содержит последовательность символов в кодировке UTF-8. Строка может быть пустой, т. е. не содержать ни одного символа. Нулевые символы в строке не допускаются.

Синтаксис

```
переменные
  п1: строка;
  п2: массив строк;
```

Ключевые слова **строка** и **строк** полностью эквивалентны и взаимозаменяемы.

Формат литерала

```
"" // пустая строка
"123абв" // последовательность символов, заключённая в кавычки
"123""абв" // для добавления в строку символа " его нужно удвоить
"\"#41№20№42#0A // последовательность 16-ричных кодов символов с кавычками в начале
"123\"#20"абв"№0A // символы в кавычках и коды можно комбинировать
```

Начальное значение

Пустая строка.

Преобразование типов

При приведении к целому или дробному числу выполняется преобразование из текстового представления числа, при ошибках возникает исключение **ОшибкаКонвертации**.

При приведении к моменту выполняется преобразование из стандартного текстового представления даты и/или времени, при ошибках возникает исключение **ОшибкаКонвертации**.

При приведении к логическому значению строка должна иметь одно из следующих значений, без учёта регистра:

```
"1", "y", "yes", "t", "true", "д", "да", "и", "истина" // истина
"0", "n", "no", "f", "false", "н", "нет", "л", "ложь" // ложь
```

Любые значения простых типов приводятся к строковому типу без ошибок.

Целое

Переменная целого типа содержит 64-битное целочисленное значение со знаком.

Синтаксис

```
переменные
  п1: целое;
  п2: массив целых;
```

Ключевые слова **целое** и **целых** полностью эквивалентны и взаимозаменяемы.

Формат литерала

```
-0123456789 // десятичные цифры
-$0123456789abcdef // 16-ричные цифры (латиница)
-§0123456789абцдеф // 16-ричные цифры (кириллица)
```

Для отрицательных чисел применяется знак «-» в начале литерала; знак «+» для положительных чисел не допускается. Регистр символов 16-ричных цифр не учитывается, латинские и кириллические цифры можно смешивать.

Начальное значение

Нуль.

Преобразование типов

Целочисленные значения приводятся к любым простым типам без ошибок, за исключением типа **объект**.

При приведении к символьному типу целое число означает кодпункт Юникод (старшие 32 бита при преобразовании отбрасываются).

При приведении к строке выполняется преобразование к десятичному текстовому представлению числа.

При приведении к моменту целое число означает количество полных суток, считая от «нулевого» момента.

При приведении к логическому истинным считается любое значение, отличное от нуля.

Дробное

Переменная дробного типа содержит 64-битное вещественное значение с двойной точностью.

Дробное число может принимать особые значения: нечисло, положительная бесконечность и отрицательная бесконечность. Такие значения нельзя указать буквально в исходном коде, но можно получить с помощью деления на нуль.

Синтаксис

```
переменные
  п1: дробное;
  п2: массив дробных;
```

Ключевые слова **дробное** и **дробных** полностью эквивалентны и взаимозаменяемы.

Формат литерала

```
-0123456.789 // представление с десятичной точкой
-0123456.789e+123 // экспоненциальное представление

д1 := 0/0; // нечисло
д2 := 1/0; // +бесконечность
д3 := -1/0; // -бесконечность
```

В качестве десятичного разделителя всегда используется точка. В качестве символа экспоненциальной части может использоваться латинская «Е» и кириллическая «Е» или «Э» в любом регистре. Если десятичная точка и экспоненциальная часть отсутствуют, литерал считается целочисленным.

Начальное значение

Нуль.

Преобразование типов

При приведении к строке число конвертируется в строковое представление в десятичном или экспоненциальном формате, в зависимости от того, какой из них короче. Для небольших чисел, если дробная часть числа равна нулю, полученное строковое представление будет целочисленным.

При приведении к строке нечисло преобразуется в текст "Nan", а бесконечность — в "+Inf" или "-Inf". Эти строковые значения (в любом регистре) можно использовать для приведения строки к дробному числу, но не в качестве литералов в исходном коде.

При приведении дробного числа к целому дробная часть отбрасывается без округления. Значение нечисло и бесконечные значения при приведении к целому вызывают исключение **НеверноеЧисло**.

При приведении к моменту число означает количество суток, считая от «нулевого» момента.

Момент

Переменная типа **момент** содержит значение даты и времени — 64-битное вещественное число с двойной точностью, означающее число суток, прошедшее от «нулевого» момента. Целая часть числа соответствует количеству полных суток, а дробная — суточному времени, выраженному в долях суток.

Нулевым моментом считается полночь 1899-12-30 00:00:00. Более поздние даты представлены положительными числами, более ранние — отрицательными.

Синтаксис

```
переменные
    п1: момент;
    п2: массив моментов;
```

Ключевые слова **момент** и **моментов** полностью эквивалентны и взаимозаменяемы.

Формат литерала

```
`2023-09-15`           // дата в обратных апострофах
`12:45:50.25`         // время в обратных апострофах
`2023-09-15 12:45`    // дата и время в обратных апострофах
`09-15 12:00`         // полдень 15 сентября текущего года
`15 12:00`           // полдень 15 числа текущего месяца
`15`                 // полночь 15 числа текущего месяца
```

Дата и время указываются в стандартном формате: **YYYY-MM-DD hh:mm:ss**.

Может быть указана только дата, только время или обе части вместе — в последнем случае дата и время разделяются пробелом. Если дата или время не указаны, они считаются равными нулю.

При указании даты можно опустить год, либо год и месяц, вместе со следующим за ними символом «-» — по умолчанию подставляются, соответственно, текущие год и месяц.

При указании времени можно опустить секунды вместе с предшествующим символом «:» — по умолчанию подставляется 0 секунд.

В качестве значения секунд можно указать дробное число.

Примечание: при указании литерала даты в исходном коде, если опущены год или месяц, то их текущие значения вычисляются в момент компиляции программы и в дальнейшем остаются постоянными.

Начальное значение

Полночь 1899-12-30.

Преобразование типов

Моменты взаимно-однозначно приводятся к дробным числам и обратно.

При приведении момента к целому числу дробная часть (т. е., значение суточного времени) отбрасывается без округления. Значение **нечисло** и бесконечные значения при приведении к целому вызывают исключение **НеверноеЧисло**.

Приведение момента к строке выполняется по следующим правилам:

- Если момент имеет значение **нечисло** или бесконечное значение, преобразование вернёт строку "Nan", "+Inf" или "-Inf".
- Моменты с ненулевой целой и дробной частью возвращают строковые представления даты и времени в стандартном формате, разделённые пробелом.
- Моменты с ненулевой целой и нулевой дробной частью возвращают строковое представление даты.
- В остальных случаях возвращается строковое представление времени.

При приведении строки к моменту допускаются такие же строковые значения, как в литералах, а также значения "nan", "inf", "+inf" или "-inf" в любом регистре. При ошибках преобразования возникает исключение **ОшибкаКонвертации**.

Логическое

Переменная логического типа имеет одно из двух значений: истинное или ложное. Внутренним представлением такого значения является 8-битное целое без знака, которое принимает значения 0 или 1.

Синтаксис

```
переменные
  p1: логическое;
  p2: массив логических;
```

Ключевые слова **логическое** и **логических** полностью эквивалентны и взаимозаменяемы.

Формат литерала

В качестве литералов в исходном коде применяются ключевые слова **да** и **нет**.

Начальное значение

Ложь.

Преобразование типов

При приведении логического значения к целому истина соответствует значению 1, а ложь — значению 0. При приведении целого к логическому истинным считается любое ненулевое значение.

При приведении к строке истина приводится к значению "да", а ложь — к значению "нет". При приведении строки к логическому допустимы следующие строковые значения, в любом регистре:

```
"1", "y", "yes", "t", "true", "д", "да", "и", "истина" // истина
"0", "n", "no", "f", "false", "н", "нет", "л", "ложь" // ложь
```

Любые другие строковые значения вызовут исключение **ОшибкаКонвертации**.

Объект

Переменная типа **объект** содержит дескриптор встроенного объекта Клаус — например, открытого файла, окна и др. подобных объектов. Такие объекты создаются и уничтожаются вызовами функций встроенной библиотеки. Переменной типа **объект** можно присвоить значение **пусто**, либо непустое значение, возвращаемое функцией. Кроме того, переменным типа **объект** можно присваивать значения других переменных этого же типа.

Значение типа **объект** имеет смысл только в пределах сеанса выполнения программы. Поэтому такие значения нельзя прочесть из файла, ввести с клавиатуры или получить любым другим способом, кроме вызова процедуры или функции, создающей объект.

Объекты, созданные при выполнении программы, должны быть уничтожены перед её завершением, иначе при завершении программы возникнет исключение **ОшибкаВыполнения**. Например, каждый открытый файл должен быть закрыт, каждое созданное окно уничтожено, и т. п.

Внимание! Присваивание переменной типа **объект** значения **пусто** не уничтожает объект. Для уничтожения объектов нужно вызывать соответствующие функции встроенной библиотеки.

Физически объект представляет собой 32-битное целое число со знаком, которое является индексом во внутреннем списке объектов Клаус. Значение **пусто** соответствует нулевому целочисленному значению, отрицательные значения не применяются.

При уничтожении объектов их целочисленные дескрипторы высвобождаются и могут быть (и, скорее всего, сразу же будут) использованы повторно при создании других объектов. Если программист присвоил значение одного и того же объекта нескольким переменным, то его ответственностью является обнуление всех этих переменных при уничтожении объекта.

Синтаксис

```
переменные
  п1: объект;
  п2: массив объектов;
```

Ключевые слова **объект** и **объектов** полностью эквивалентны и взаимозаменяемы.

Формат литерала

В качестве литерала в исходном коде применяется ключевое слово **пусто**. Иные значения дескриптора объекта не могут быть указаны буквально.

Начальное значение

Пусто.

Преобразование типов

Приведение других типов к типу **объект** не допускается.

Приведение значения типа **объект** к целому возвращает физическое значение дескриптора — 32-битное целое число. Значение **пусто** приводится к нулю.

При приведении объекта к строке возвращается строковое представление численного значения дескриптора в 16-ричном формате.

Составные типы

Массив

Переменная типа **массив** содержит несколько однотипных значений, каждому из которых соответствует целочисленный индекс. Первый элемент массива имеет индекс 0, последний элемент — индекс на единицу меньше, чем длина массива. Индексы в массиве следуют один за другим, без пропусков. Элементы массива могут иметь любой тип — простой или составной.

Синтаксис

Определение массивного типа состоит из ключевого слова **массив** и следующего за ним определения типа элемента массива. Для определения многомерных массивов можно использовать разные формы ключевого слова: **массив массивов**, **массив массивов массивов** и т. п.

```
тип
    тМЦ = массив целых;
    тМСДкС = массив словарей дробных ключ символ;
переменная
    мойМассив: массив массивов моментов;
    ещёМассив: массив словарей структур
        поле1: символ;
        поле2: строка;
    окончание ключ строка;
```

Ключевые слова **массив** и **массивов** полностью эквивалентны и взаимозаменяемы.

В исходном коде программы для обращения к элементу массива нужно указать индекс элемента в квадратных скобках после имени переменной-массива. Для работы с массивами и их элементами используются функции **длина ()**, **добавить ()**, **вставить ()**, **удалить ()** и конструкции **есть ()**, **нету ()**.

```
переменная
    м1: массив целых;
    м2: массив массивов моментов;
начало
    длина (м1, 1);
    м1 [0] := 10;
    длина (м2, 2);
    длина (м2 [1], 11);
    м2 [1] [10] := `2023-07-06 12:00`;
окончание.
```

Начальное значение

Пустой массив.

Преобразование типов

Явное приведение массивов к другим типам данных запрещено.

Переменной типа **массив** можно присвоить значение — массив, состоящий из элементов такого же типа или типа, совместимого по присваиванию. Например, массиву дробных можно присвоить массив целых, а массиву строк можно присвоить массив символов.

Словарь

Переменная типа **словарь** содержит несколько однотипных значений, каждому из которых соответствует значение ключа, уникальное в пределах словаря. Ключи в словаре отсортированы в порядке возрастания. Элементы словаря могут иметь любой тип — простой или составной, а ключи — любой из простых типов.

Синтаксис

Определение словарного типа состоит из ключевого слова **словарь** и следующего за ним определения типа элемента словаря; после этого следует слово **ключ** и ключевое слово, указывающее тип ключа. Затем, если ключ словаря имеет тип **дробное** или **момент**, может следовать ключевое слово **точность**, а за ним константное выражение дробного типа, определяющее точность сравнения ключей.

Определением типа элемента словаря может служить ключевое слово, идентификатор ранее объявленного типа, либо безымянное определение типа. Определением типа ключа может быть только ключевое слово простого типа данных.

```
тип
    тСЦ = словарь целых ключ строка;
    тСМДкС = словарь массивов дробных ключ символ;
переменная
    мойСловарь: словарь моментов ключ объект;
    ещёСловарь: словарь структур
        поле1: символ;
        поле2: строка;
    окончание ключ строка;
```

Ключевые слова **словарь** и **словарей** полностью эквивалентны и взаимозаменяемы.

В исходном коде программы для обращения к элементу словаря нужно указать ключ элемента в квадратных скобках после имени переменной-словаря. Для добавления в словарь пары ключ-значение можно использовать инструкцию присваивания. Кроме того, для работы со словарями и их элементами используются функции **длина ()**, **добавить ()**, **вставить ()**, **удалить ()** и конструкции **есть ()**, **нету ()**.

```
переменная
    сл1: словарь целых ключ символ;
    сл2: словарь словарей моментов ключ целое ключ строка;
начало
    сл1['0'] := 10;
    сл2["первый"][10] := `2023-07-06 12:00`;
окончание.
```

Для словарей с ключом типа **дробное** или **момент** можно указать точность сравнения ключей. Если указана ненулевая точность, то два ключевых значения будут считаться равными, если модуль их разности не превышает указанного значения точности.

```
переменная
    сл: словарь строк ключ дробное точность 0.01;
начало
    сл[0] := "Нуль";
    сл[0.001] := "Снова нуль";
    сл[0.02] := "Две сотых";
окончание.
```

В приведённом примере в словарь будут добавлены два значения с ключами 0 и 0.02, т. к. ключевое значение 0.001 совпадёт с нулём в пределах указанной точности, и строковое значение "Снова нуль" будет присвоено ранее добавленному нулевому ключу.

Начальное значение

Пустой словарь.

Преобразование типов

Явное приведение словарей к другим типам данных запрещено.

Переменной типа **словарь** можно присвоить значение — словарь, состоящий из элементов такого же типа или типа, совместимого по присваиванию, при условии точного совпадения типов ключей. Например, словарю дробных можно присвоить словарь целых, а словарю строк можно присвоить словарь символов.

Структура

Переменная типа **структура** содержит одно или несколько именованных значений различных типов, называемых полями структуры; значением структурной переменной является совокупность значений всех её полей. Имена полей должны быть уникальны в пределах структуры. Определение структуры является статическим, т. е. состав её полей не может быть изменён в процессе выполнения программы.

Синтаксис

Определение структурного типа начинается ключевым словом **структура** и заканчивается ключевым словом **окончание**, между которыми располагаются определения полей структуры. Определения полей синтаксически идентичны определениям переменных, с той оговоркой, что для полей структур не могут быть указаны начальные значения.

Типы полей структур могут быть простыми или составными; при объявлении полей допускаются безымянные определения типов, в т.ч. определения вложенных структур. Имена полей вложенной структуры должны быть уникальны в её пределах, но могут совпадать с именами полей объемлющей структуры.

Ключевые слова **структура** и **структур** полностью эквивалентны и взаимозаменяемы.

```
тип
    тСтр = структура
        п1: целое;
        п2, п3: дробное;
        п4: массив строк;
        п5: структура
            п1: символ;
            п2: словарь объектов ключ строка;
        конец;
    окончание;
переменная
    точка: структура х, у: целое; конец;
    люди: массив структур
        фамилия, имя, отчество: строка;
        ДР: момент;
    конец;
```

В исходном коде программы для обращения к полю структуры нужно указать имя поля, отделив его от имени переменной знаком «.». Аналогичным образом можно обращаться к полям вложенных структур.

```
переменная
    стр: структура
        п1: строка;
        п2: структура
            п1, п2: целое;
        конец;
    окончание;
```

```
начало
    стр.п1 := "значение";
    стр.п2.п1 := 10;
окончание.
```

Начальное значение

Все поля структуры имеют начальные значения, соответствующие их типам.

Преобразование типов

Явное приведение структур к другим типам данных запрещено.

Переменной типа **структура** можно присвоить значение — структуру, имеющую точно такое же количество полей с точно такими же именами (без учёта регистра) и типами, совместимыми по присваиванию. Например, структуре, имеющей одно строковое поле, можно присвоить структуру, имеющую одно символьное поле с таким же именем.

Выражения и операции

Выражения в Клаусе записываются в инфиксной форме, для группировки операций используются круглые скобки.

Операции в выражениях выполняются в порядке возрастания приоритета (чем меньше приоритет, тем раньше — см. табл. ниже). Равноприоритетные операции выполняются по порядку слева направо. Для бинарных операций сначала вычисляется левый, а затем правый операнд. Все операнды выражений обязательно вычисляются и все операции над ними выполняются, даже если это излишне для вычисления итогового значения выражения.

Операции сложения, вычитания и умножения возвращают целый результат, если оба операнда целые. Если хотя бы один операнд дробный, то операция вернёт дробный результат. Операция деления всегда возвращает дробный результат.

Операции целочисленного деления и остатка от деления, а также побитовые операции определены только для целочисленных операндов и возвращают целочисленный результат.

Сложение и вычитание определены для чисел, либо для момента и числа: можно прибавить к моменту (целое или дробное) число суток и вычесть из момента число суток. В обоих случаях момент должен быть левым операндом, и результатом операции является момент.

Операция соединения определена для символов и строк в любых сочетаниях и возвращает строковый результат. При этом нулевые символы приводятся к пустой строке.

Операции сравнения определены для всех простых типов, сравнимых между собой:

- символы можно сравнивать с символами, но не со строками;
- строки можно сравнивать со строками, но не с символами;
- целые и дробные числа можно сравнивать между собой в любых сочетаниях;
- моменты можно сравнивать с моментами;
- логические значения можно сравнивать между собой: **да** больше, чем **нет**.
- объекты можно сравнивать между собой, но это не имеет иного смысла, кроме возможности создать словарь с ключом типа **объект**. При сравнении объектов значение **пусто** считается наименьшим.

Операции сравнения запрещены для дробных чисел и моментов, имеющих значение **нечисло**. Если хотя бы один из аргументов операции сравнения является нечислом, возникает исключение **НеверноеЧисло**.

Арифметические операции для нечисел разрешены, но если хотя бы один из аргументов является нечислом, то результатом операции также будет **нечисло**.

Значения составных типов — массивы, словари, структуры — не могут быть операндами в выражениях.

Таблица операций

Знак	Операция	Операнды	Результат	Приоритет
-	Унарный минус	число	число	0
!	Унарное отрицание	логический, целый	логический, целый	0
^	Возведение в степень	числа	число	1
*	Умножение	числа	число	2
/	Деление	числа	число	2
\	Целочисленное деление	целые	целое	2
%	Остаток от деления	целые	целое	2
+	Сложение	числа, момент	число, момент	3
-	Вычитание	числа, момент	число, момент	3
&	Побитовое И	целые	целое	4
	Побитовое ИЛИ	целые	целое	5
~	Побитовое Искл. ИЛИ	целые	целое	5
++	Соединение	строки, символы	строка	6
=	Равно	простые	логический	7
<>	Не равно	простые	логический	7
>	Больше	простые	логический	7
<	Меньше	простые	логический	7
>=	Больше или равно	простые	логический	7
<=	Меньше или равно	простые	логический	7
&&	Логическое И	логические	логический	8
	Логическое ИЛИ	логические	логический	9
~~	Логическое Искл. ИЛИ	логические	логический	9

Инструкции

Знак «;»

После каждой инструкции в языке Клаус требуется знак «;» (точка с запятой). Этот знак является ни частью инструкции, ни разделителем между инструкциями, но признаком окончания инструкции.

Знак «;» не допускается:

- в условном ветвлении перед ключевым словом **иначе**, если оно есть;
- в цикле с постусловием перед ключевым словом **пока**.

Во всех остальных случаях знак «;» требуется после каждой инструкции, за исключением окончания тела основной программы, где требуется знак «.» (точка).

Инструкция присваивания

Инструкция присваивания предназначена для установки значения переменной или части переменной — элемента массива, элемента словаря, поля структуры.

В Клаусе определены две разновидности инструкций присваивания: простая и вычисляющие.

Простое присваивание

Простое присваивание записывается в следующей форме:

```
<переменная> := <выражение>;
```

где **переменная** — путь к переменной, а **выражение** должно возвращать значение совместимого типа, которое будет присвоено указанной переменной.

Путь к переменной состоит из имени переменной и последовательности квалификаторов — индексов массивов, ключей словарей и имён полей структур.

Для присваивания значения полю структуры имя поля указывается через точку после пути к структуре. Для присваивания значения элементу массива (словаря) индекс массива (ключ словаря) указывается в квадратных скобках после пути к массиву/словарю, например:

```
программа Пример;
типы
    тСтр1 = структура
        с1п1: строка;
        с1п2: дробное;
    окончание;
    тСтр2 = структура
        с2п1: массив массивов тСтр1;
        с2п2: строка;
    окончание;
переменные
    а: целое;
    б: дробное;
    в: массив строк;
    г: словарь строк ключ дробное;
    сл: словарь тСтр2 ключ строка;
    м: массив тСтр1;
начало
    а := 10;
    б := а + 0.01;
    длина(в, 1);
    в[0] := "значение";
    г[3.1415] := "Пи";
    добавить(сл, "Ключ");
    длина(сл["Ключ"].с2п1, 2);
    сл["Ключ"].с2п1[0] := м;
    длина(сл["Ключ"].с2п1[1], 1);
    сл["Ключ"].с2п1[1][0].с1п1 := "Новое значение";
окончание.
```

Вычисляющее присваивание

В Клаусе определены следующие инструкции вычисляющего присваивания:

+=	Прибавить и присвоить
-=	Вычесть и присвоить
*=	Умножить и присвоить
/=	Разделить и присвоить
\=	Разделить нацело и присвоить
%=	Разделить и присвоить остаток от деления
^=	Возвести в степень и присвоить
&=	Вычислить побитовое «И» и присвоить
 =	Вычислить побитовое «ИЛИ» и присвоить
~=	Вычислить побитовое «Исключающее ИЛИ» и присвоить

Каждая из этих инструкций выполняет арифметическую или побитовую операцию, используя в качестве левого операнда значение переменной, указанной слева, а в качестве правого — значение выражения, указанного справа. Результат операции становится новым значением переменной, например:

```
начало
  A += 1; // прибавляет к переменной A единицу
  B *= 2; // удваивает значение переменной B
  B ^= 3; // возводит переменную B в третью степень
  B |= 8; // устанавливает единицу в 3-м (считая от 0) бите переменной B
окончание;
```

Совместимость типов при присваивании

Тип значения, возвращаемый выражением в правой части инструкции, должен быть совместим по присваиванию с переменной в левой части.

Для переменных простых типов:

- Строковой переменной может быть присвоено символьное значение, но не наоборот;
- Дробной переменной может быть присвоено целое значение, но не наоборот;
- В остальных случаях требуется точное соответствие типов.

Для переменных составных типов:

- Массиву может быть присвоен массив, состоящий из элементов совместимого типа;
- Словарю может быть присвоен словарь, имеющий ключ точно такого же типа и состоящий из элементов совместимого типа.
- Структуре может быть присвоена структура, имеющая ровно столько же полей с точно такими же именами (без учёта регистра) и значениями совместимых типов. Порядок следования полей не имеет значения.

Таким образом, например, массиву дробных чисел можно присвоить массив целых чисел, а структуре, имеющей строковое поле, можно присвоить структуру, имеющую одноимённое символьное поле.

Копирование данных при присваивании

С логической точки зрения, каждая переменная языка Клаус содержит свою собственную копию данных. При этом гарантируется, что присваивание значения одной переменной никогда не повлечет за собой изменение значения какой-нибудь другой переменной.

Физически, при присваивании значений переменных, а также при передаче параметров в процедуры и функции, используется механизм копирования-по-требованию. А именно, при присваивании значения переменная-приёмник получает ссылку на ту же копию данных, что хранится в переменной-источнике, так что в оперативной памяти имеется лишь одна копия данных, на которую ссылаются две (или более) переменных.

Далее, если значение одной из переменных модифицируется, то перед этим для неё создаётся уникальная копия данных, и все изменения вносятся в эту вновь созданную копию. Таким образом, копирование данных в оперативной памяти выполняется только тогда, когда оно в самом деле необходимо.

Нужно учитывать, что копирование-по-требованию работает только для переменных, имеющих полностью идентичные типы данных. Например, присваивание словаря целых словарю дробных вызовет немедленное копирование всех ключей и значений из одного словаря в другой.

Кроме того, копирование-по-требованию выполняется лишь для всего значения переменной как целого и не работает ни для отдельных элементов массивов и словарей, ни для отдельных полей структур. Например, если

переменная A содержит ссылку на массив, полученную от переменной B, и первому элементу этого массива присваивается новое значение, то перед выполнением присваивания для переменной A будет создана новая копия всего массива.

Инструкция вызова подпрограммы

Для вызова подпрограммы — процедуры или функции — в исходном коде записывается её имя, за которым должны следовать круглые скобки. Если для подпрограммы определены параметры, то перечень значений параметров указывается в скобках через запятую в том количестве и в том порядке, как это предусмотрено определением подпрограммы. Если параметры отсутствуют, круглые скобки должны быть пустыми.

```
начало
    мояПроцедура ();
    мояВтораяПроцедура (10, 20, "тридцать");
окончание;
```

Таким образом можно вызывать как процедуры, так и функции — в последнем случае значение, возвращаемое функцией, теряется. Если возвращаемое функцией значение нужно использовать в дальнейшем по ходу программы, то его необходимо присвоить некоторой переменной, например:

```
начало
    мояПеременная := мояФункция ('А', "Б");
окончание;
```

Функции, возвращающие значения простых типов можно, кроме того, использовать в качестве аргументов в выражениях:

```
начало
    результат := вычислитьОдно (1, 2) + вычислитьДругое (3, 4, 5) + 1;
    если вычислитьТретье () > 0 то результат := 0;
окончание;
```

Ничего — пустая инструкция

Пустая инструкция записывается ключевым словом **ничего**:

```
начало
    ничего;
окончание;
```

Эта инструкция не выполняет никаких действий. Она может использоваться в тех случаях, когда правила синтаксиса требуют наличия инструкции, которая фактически не нужна, например:

```
начало
    если ВсёХорошо то ничего
    иначе сообщить ("Что-то пошло не так!");
окончание;
```

Завершить — завершение программы

Для завершения работы программы используется инструкция **завершить**:

```
завершить;  
завершить 0;  
завершить кодОшибки;
```

где **кодОшибки** — целое число (или выражение целочисленного типа), которое будет выдано операционной системе в качестве кода завершения процесса. Если программа была выполнена успешно, она должна вернуть код завершения равный нулю. Ненулевой код завершения свидетельствует об ошибке, возникшей при выполнении программы.

Так, если при выполнении программы возникают необработанные исключения, программа будет по умолчанию завершена с кодом -1. Если все инструкции, составляющие тело программы, были выполнены успешно, и программа не была явно завершена инструкцией **завершить**, то код завершения по умолчанию будет равен нулю. Аналогично, инструкция **завершить** вызывает успешное завершение программы, если код не указан явно.

Инструкция **завершить** допустима внутри подпрограмм любого уровня вложенности. Если инструкция **завершить** выполняется внутри составного блока, содержащего секцию **напоследок**, то инструкции в этой секции будут выполнены перед выходом из блока — так, как если бы инструкция **завершить** вызвала исключение.

Вернуть — возврат из подпрограммы

Инструкция возврата используется для возврата из процедуры или функции с указанием значения, возвращаемого функцией.

Инструкция записывается ключевым словом **вернуть**, после которого может быть указано возвращаемое значение функции.

```
функция мояФункция(): строка;  
начало  
    вернуть "результат";  
окончание;  
  
процедура мояПроцедура();  
начало  
    вернуться;  
окончание;
```

Ключевые слова **вернуть** и **вернуться** полностью эквивалентны и взаимозаменяемы.

Если инструкция используется в теле функции, то после ключевого слова должно быть указано значение (или выражение), возвращаемое функцией. Если все инструкции в теле функции были выполнены, и функция не была явно завершена инструкцией **вернуть**, то она вернёт нулевое начальное значение соответствующего типа.

Для процедур указывать возвращаемое значение не допускается.

Если инструкция **вернуть** выполняется в теле основной программы, то она работает в точности так же, как инструкция **завершить**. В этом случае возвращаемое значение должно быть целым числом и означает код завершения процесса.

Если инструкция **вернуть** выполняется внутри составного блока, содержащего секцию **напоследок**, то инструкции в этой секции будут выполнены перед выходом из блока — так, как если бы инструкция **вернуть** вызвала исключение.

Начало-конец — составной блок инструкций

Составной блок представляет собой инструкцию, которая может содержать одну или несколько других инструкций и предоставляет средства для обработки исключений.

Составной блок начинается ключевым словом **начало** и заканчивается словом **конец** или **окончание**. Ключевые слова **конец** и **окончание** полностью эквивалентны и взаимозаменяемы. Между началом и окончанием блока может находиться одна или несколько других инструкций (в т. ч., вложенные составные блоки).

```
начало
    выполнитьОдно ();
    выполнитьДругое ();
конец;
```

Составной блок может применяться во всех случаях, когда синтаксис языка требует наличия ровно одной инструкции, а фактически их требуется несколько — в условных ветвлениях, в циклах, в инструкции выбора и др. Тело основной программы и любой процедуры или функции также представляет собой составной блок.

```
если А > В то чтоТоОдно ()
иначе чтоТоДругое ();

если Ы < Я то начало
    чтоТоОдно ();
    второе ();
    третье ();
конец иначе начало
    чтоТоДругое ();
    четвертое ();
    пятое ();
конец;
```

Инструкции в теле составного блока выполняются по порядку одна за другой. Если при выполнении некоторой инструкции возникло исключение, то следующие за ней инструкции не выполняются, и управление немедленно передаётся в секцию обработки исключений.

Секция обработки исключений, если она есть в блоке, располагается после всех инструкций, составляющих блок, и начинается ключевым словом **исключение** или **исключения**.

```
начало
    выполнитьОдно ();
    выполнитьВторое ();
    выполнитьТретье ();
исключение
    сообщить ("Возникла какая-то ошибка!");
окончание;
```

Если блок не имеет секции обработки исключений, то возникшее исключение всплывает в объемлющий блок, имеющий такую секцию — в т. ч., возможно, за пределы выполняемой подпрограммы. Если такой объемлющий блок вообще отсутствует, то выполнение программы завершается с кодом -1.

Секция обработки исключений может быть простой или детализированной.

Простая секция содержит одну или несколько любых инструкций. Если блок имеет простую секцию обработки исключений, то при возникновении в блоке любого исключения управление передаётся в эту секцию, имеющиеся в ней инструкции выполняются, и любое исключение считается подавленным, если его всплытие не было возобновлено.

Чтобы возобновить всплытие исключения, в секции обработки исключений (и только в ней) можно использовать инструкцию **бросить**.

Детализированная секция обработки исключений содержит одну или несколько инструкций перехвата исключений **когда...тогда** и не может содержать никаких других инструкций. Исключение, попавшее в такую секцию, считается подавленным в том случае, если оно соответствует какой-либо из инструкций перехвата; все остальные исключения всплывают в объемлющий блок.

```
начало
  ч := целое ("Не число");
исключение
  когда ОшибкаКонвертации тогда ч := 0;
  когда любое тогда сообщить ("Какая-то неожиданная ошибка!");
конец;
```

Составной блок может заканчиваться секцией гарантированного завершения.

Если в блоке есть секция гарантированного завершения, то перед выходом из блока — вне зависимости от того, возникло ли в нём исключение, или все инструкции были выполнены успешно — управление передаётся в эту секцию. Также управление передаётся в эту секцию при выходе из блока инструкциями **прервать**, **продолжить**, **вернуть** и **завершить**.

Секция гарантированного завершения начинается ключевым словом **напоследок**. Эта секция должна располагаться в конце блока, после основной секции и секции обработки исключений.

```
скрытьКурсор ();
начало
  чтоТоВыполнить ();
  ещёЧтоТоВыполнить ();
  вернуть 0;
исключение
  сообщить (текст_исключения ());
напоследок
  показатьКурсор ();
конец;
```

Когда-тогда — перехват исключений

Инструкция перехвата исключения может использоваться только в секции обработки исключений составного блока (т. е., после ключевого слова **исключение**) и не может соседствовать в ней с какими-то ещё инструкциями, кроме других инструкций перехвата.

Инструкция начинается ключевым словом **когда**, за которым следует одно из трёх:

- одно или несколько имён исключений через запятую;
- определение переменной исключения;
- ключевое слово **любое** (или **любой**, **любая**).

Далее следует ключевое слово **тогда**, а за ним тело инструкции перехвата — ровно одна инструкция (в т. ч., возможно, составной блок), которая будет выполнена в том случае, если подходящее исключение возникло при выполнении основной секции составного блока.

Чтобы просто подавить исключение, телом инструкции перехвата может быть инструкция **ничего**. Чтобы возобновить всплытие перехваченного исключения, используется инструкция **бросить**.

```

начало
    выполнитьОдно ();
    выполнитьДругое ();
исключения
    когда ОшибкаКонвертации тогда сообщить ("Неверный формат данных.");
    когда НеверныйИндекс, НеверныйКлюч тогда сообщить ("Неверный программист.");
    когда ош: МояОшибка тогда начало
        ош.текст := ош.текст ++ " Не волнуйтесь, так и было задумано! ;)";
        бросить;
    конец;
    когда любое тогда завершить 100;
конец;

```

Запрещено упоминать одно и то же исключение в нескольких инструкциях перехвата в пределах одной секции. Инструкция **когда любое** перехватывает любое возникшее исключение, если оно не упомянуто в других инструкциях в той же секции. Такая инструкция должна быть последней в секции.

Если в инструкции перехвата не объявлена переменная, а перечислены одно или несколько имён исключений, то тело инструкции выполняется при возникновении любого из них. При этом можно узнать, какое именно из исключений возникло, и получить текст сообщения об ошибке вызовом функций **имя_исключения ()** и **текст_исключения ()**, но нет возможности модифицировать текст или параметры исключения.

Для доступа к параметрам исключения нужно объявить переменную — указать её имя после ключевого слова **когда** и отделить двоеточием от имени исключения. Такая инструкция может перехватывать только один вид исключения.

Переменная представляет собой структуру, которая содержит:

- Поле **текст** строкового типа, содержащее текст сообщения об ошибке;
- Поля простых типов, соответствующие параметрам в определении исключения.

Эти поля получают свои значения при создании исключения и могут быть изменены при его перехвате. Следует иметь в виду, что если текст сообщения об ошибке содержал спецификаторы вида **% (имя)**, то в момент перехвата поле **текст** содержит уже отформатированный текст, где все спецификаторы заменены на значения параметров. Изменение значений параметров исключения при его перехвате не приводит к повторному формированию текста сообщения об ошибке; для того, чтобы изменить текст сообщения, нужно присвоить новое значение полю **текст**.

Модификация текста и параметров перехваченного исключения имеет смысл лишь в том случае, если его всплытие будет возобновлено инструкцией **бросить**.

Ошибка — создание исключения

Эта инструкция применяется для того, чтобы прервать последовательность выполнения и создать всплывающее исключение.

Инструкция начинается ключевым словом **ошибка**, за которым следует имя исключения. Если для исключения определены параметры, то их значения должны быть перечислены по порядку в круглых скобках после имени исключения. Далее может следовать ключевое слово **сообщение**, а за ним выражение строкового типа — текст сообщения об ошибке.

Инструкции, следующие после инструкции **ошибка**, не выполняются. Вместо этого управление передаётся в секцию обработки исключений ближайшего составного блока. Если такой блок отсутствует, выполнение программы будет завершено с кодом -1.

```

исключение
    МояОшибка (п1, п2: целое; п3: строка) сообщение "Тут какая-то ошибка";
    ЕщёОшибка (стр, симв: целое);

```

```
начало
    ошибка ОшибкаВыполнения;
    ошибка МояОшибка(10, 20, "тридцать");
    ошибка ЕщёОшибка(1, 2) сообщение "Ошибка в строке %(стр), символ %(симв).";
конец;
```

Если текст сообщения об ошибке был указан и в определении исключения, и при его создании, то последний имеет приоритет. Если текст не указан ни в определении, ни при создании исключения, то в качестве текста сообщения используется имя исключения.

Если для исключения определены параметры, то текст сообщения об ошибке может содержать спецификаторы вида `% (имя)`, где **имя** — имя параметра исключения. При создании исключения такие спецификаторы будут заменены на строковые представления значений соответствующих параметров.

Кроме того, значения параметров исключения могут быть получены в секции перехвата исключений составного блока, как это описано в разделе **Инструкция перехвата исключения**.

Бросить — воссоздание исключения

Эта инструкция записывается ключевым словом **бросить**. Её можно применять только в секции обработки исключений составного блока: она прерывает последовательность выполнения и воссоздаёт то же самое исключение, которое привело к передаче управления в данную секцию обработки исключений.

Инструкции, следующие после инструкции **бросить**, не выполняются. Вместо этого управление передаётся в секцию обработки исключений, расположенную в ближайшем объёмлющем составном блоке. Если такой блок отсутствует, выполнение программы будет завершено с кодом -1.

Если параметры исключения или текст сообщения об ошибке были модифицированы в секции обработки исключений, то исключение будет воссоздано с новыми значениями.

Если-то-иначе — условное ветвление

Эта инструкция применяется для выполнения тех или иных действий в зависимости от некоторого условия.

Инструкция начинается ключевым словом **если**, после которого должно быть указано условие — значение или выражение логического типа. Далее следует ключевое слово **то**, а за ним — ровно одна инструкция, которая будет выполнена в том случае, если указанное условие имеет истинное значение.

Далее может следовать знак «;», либо ключевое слово **иначе**, а за ним ещё ровно одна инструкция и знак «;». Эта последняя инструкция будет выполнена в том случае, когда указанное условие имеет ложное значение.

В обоих случаях ровно одна инструкция может представлять собой составной блок, содержащий сколько угодно инструкций.

```
начало
    если А < В то
        вывести ("А меньше В");
    если Н < М то
        вывести ("Н меньше М")
    иначе начало
        вывести ("Н больше или равно М");
        если Н > М то вывести ("Всё-таки больше");
    конец;
конец;
```

Если требуется перебрать несколько условий подряд, то инструкции ветвления можно располагать друг за другом:

```
начало
    если имя = "Петя" то вывести("Узнаю брата Петю!")
    иначе если имя = "Вася" то вывести("Узнаю брата Васю!")
    иначе если имя = "Коля" то вывести("Узнаю брата Колю!")
    иначе вывести("Что-то не узнаю...");
конец;
```

Здесь действует правило: каждое **иначе** относится к ближайшему **если**, расположенному выше по тексту исходного кода.

Во многих подобных случаях будет удобнее воспользоваться инструкцией многовариантного выбора.

Выбор-из — многовариантный выбор

Эта инструкция применяется для выполнения тех или иных действий в зависимости от выбора значения из нескольких предусмотренных вариантов.

Инструкция начинается ключевым словом **выбор**, за которым следует выбираемое значение — буквальное значение или выражение простого типа. Затем, если выбираемое значение имеет тип **дробное** или **момент**, может следовать ключевое слово **точность**, а за ним константное выражение дробного типа, определяющее точность сравнения. Далее следует слово **из**, а за ним тело инструкции и слово **окончание**.

Тело инструкции состоит из одного или нескольких описателей варианта выбора. Каждый описатель начинается с перечня подходящих значений, перечисленных через запятую, далее следует знак «:», а за ним ровно одна инструкция и знак «;». Эта инструкция будет выполнена в том случае, если данный вариант выбора окажется подходящим. Она может представлять собой составной блок, содержащий сколько угодно инструкций.

Перечень подходящих значений может состоять из буквальных значений или константных выражений — т. е., в качестве операндов этих выражений могут использоваться только буквальное значения и константы, а переменные и вызовы функций не допускаются. Тип перечисленных значений должен точно соответствовать типу выбираемого значения.

Вариант выбора считается подходящим, если выбираемое значение точно совпадает с одним из значений, перечисленных в описателе варианта (дробные значения и моменты должны совпадать с указанной точностью). Тело инструкции выбора не может содержать нескольких описателей с совпадающими значениями.

В последнем из описателей вместо перечня подходящих значений можно указать ключевое слово **иначе** (без двоеточия). Такой описатель будет считаться подходящим в том случае, если выше не нашлось никакого другого подходящего описателя.

```
программа Пример;
переменная
    имя: строка;
начало
    сообщить("Введите имя: ");
    ввести(имя);
    выбор имя из
        "Петя": вывести("Узнаю брата Петю!");
        "Вася": вывести("Узнаю брата Васю!");
        "Коля": вывести("Узнаю брата Колю!");
        "Оля", "Даша", "Вика": вывести("Узнаю сестрёнку!");
        "Таня": вывести("Привет, Таня :)");
        иначе вывести("Что-то не узнаю...");
    конец;
окончание.
```

```

программа ЕщёПример;
переменная
  N: дробное = 0.345;
начало
  выбор N точность 0.09 из
    0: вывести("Нуль");
    0.1: вывести("Одна десятая");
    0.2: вывести("Две десятых");
    0.3: вывести("Три десятых"); // совпадёт с указанной точностью
    0.4: вывести("Четыре десятых");
  конец;
окончание.

```

Для выбора из значений типа **дробное** или **момент** можно указать точность сравнения. Если указана ненулевая точность, то вариант выбора будет считаться подходящим, если модуль разности указанной константы и выбираемого значения не превышает указанного значения точности.

Примечание: следует быть осторожным при использовании этой инструкции с дробными числами и моментами без указания точности. Сравнение вещественных чисел на точное равенство может давать непредсказуемый результат в зависимости от операционной системы и аппаратного обеспечения.

Для-от-до-цикл — цикл по счётчику

Эта инструкция предназначена для выполнения одних и тех же действий несколько раз. Количество повторений (итераций) цикла вычисляется заранее.

Инструкция начинается ключевым словом **для**, за которым следует имя переменной — счётчика цикла. Далее следует слово **от**, за ним выражение, определяющее начальное значение счётчика, а затем слово **до** и выражение, определяющее конечное значение счётчика. После этого следует слово **цикл** и ровно одна инструкция, составляющая тело цикла. Тело цикла может быть составным блоком, содержащим сколько угодно инструкций. Перед словом **цикл** допускается слово **обратный**.

Счётчиком цикла может быть только локальная переменная целого типа — она должна быть объявлена в разделе определений той программы или подпрограммы, в теле которой выполняется цикл. Выражения начального и конечного значения счётчика также должны возвращать целое. Значения этих выражений вычисляются один раз перед первой итерацией цикла. Изменение внутри цикла каких-либо данных, влияющих на значения этих выражений, не приведёт к изменению количества итераций.

```

программа Пример;
переменная
  цвета: массив целых;
  ц: целое;
начало
  длина(цвета, 16);
  для ц от 0 до 15 цикл цвета[ц] := ц;
  для ц от длина(цвета)-1 до 0 обратный цикл начало
    цветФона(цвета[ц]);
    вывести("Цв", строка(ц));
  конец;
окончание.

```

Перед первой итерацией счётчику цикла присваивается начальное значение. Затем выполняется тело цикла. После этого значение счётчика увеличивается на 1, и тело цикла выполняется снова. Так происходит до тех пор, пока значение счётчика меньше или равно конечному значению. После этого цикл завершается, и начинают выполняться инструкции, следующие после цикла.

Значение переменной-счётчика цикла можно менять в теле цикла, но это не приведёт к изменению количества итераций. Переменная-счётчик сохранит присвоенное ей значение до конца текущей итерации, после чего ей

будет присвоено следующее заранее вычисленное значение, так что на входе в следующую итерацию цикла счётчик будет иметь значение на 1 больше, чем на входе в нынешнюю.

Если начальное значение счётчика больше конечного, тело цикла не будет выполнено ни разу.

Если перед ключевым словом **цикл** указано слово **обратный**, то цикл выполняется в обратном порядке: при каждой итерации значение счётчика уменьшается на единицу. В этом случае начальное значение счётчика должно быть больше или равно конечному.

Выполнение цикла можно прервать инструкцией **прервать**. Инструкция **продолжить** прерывает текущую итерацию цикла и немедленно начинает следующую — при том условии, что счётчик ещё не равен конечному значению.

Для-каждого-цикл — цикл по набору данных

Эта инструкция предназначена для выполнения одних и тех же действий для каждого элемента из набора данных. Набором данных может служить массив, словарь или строка — в последнем случае элементом набора данных является каждый символ в строке.

Инструкция начинается ключевыми словами **для каждого**, за которыми следует имя переменной-итератора. Переменная-итератор должна быть объявлена в локальной области видимости — т. е., в разделе определений той программы или подпрограммы, в теле которой выполняется цикл.

Далее следует ключевое слово **из** и имя переменной, содержащей набор данных — строку, массив или словарь. Перед словом **из** допустимо слово **от**, а за ним выражение, определяющее стартовую позицию итератора в наборе данных.

Затем следует слово **цикл** и ровно одна инструкция, составляющая тело цикла. Тело цикла может быть составным блоком, содержащим сколько угодно инструкций. Перед словом **цикл** допускается слово **обратный**.

Для массивов и строк итератор должен быть целочисленного типа. Для массива он означает индекс элемента в массиве, а для строки — номер байта в строке, с которого начинается очередной символ. Байты в строке и элементы в массиве нумеруются начиная с нуля.

Для словарей тип итератора должен точно соответствовать типу ключа словаря — при каждой итерации цикла этой переменной присваивается очередное значение ключа.

```
программа ЦиклПоСтроке;
переменные
    б: целое;
    с: строка;
    смв: символ;
начало
    сообщить ("Введите строку: ");
    ввести (с);
    для каждого б из с цикл начало
        смв := симв (с, б);
        вывести (б, ": ", смв, "\n");
    конец;
окончание.
```

```
программа ЦиклПоМассиву;
переменные
    м: массив строк;
    кво, и: целое;
начало
    сообщить ("Введите количество: ");
    ввести (кво);
    длина (м, кво);
```

```
для каждого и из м цикл начало
    сообщить ("Введите элемент ", и, ": ");
    ввести(м[и]);
конец;
сообщить ("Значения в обратном порядке:", НС);
для каждого и из м обратный цикл вывести(м[и], НС);
окончание.
```

программа ЦиклПоСловарю;

переменные

с: строка;

сл: словарь строк ключ строка;

начало

сл["Иванов"] := "Иванов Иван Иванович";

сл["Петров"] := "Петров Пётр Петрович";

сл["Сидоров"] := "Сидоров Сидор Сидорович";

сл["Николаев"] := "Николаев Николай Николаевич";

сл["Захаров"] := "Захаров Захар Захарович";

для каждой с из сл от "Н" цикл вывести(сл[с], НС);

окончание.

Выражение, следующее за ключевым словом **от**, определяет тот элемент набора данных, с которого начнётся цикл. Тип значения этого выражения должен точно соответствовать типу переменной-итератора.

- Для строки эта возможность не реализована в нынешней версии языка.
- Для массива — выражение должно возвращать целое число, означающее индекс элемента массива, с которого начнётся цикл.
- Если в словаре есть ключ, равный значению выражения, то цикл начнётся с этого ключа. Если такого ключа нет, то цикл начнётся с ближайшего следующего значения ключа: прямой цикл с ближайшего большего значения, а обратный цикл с ближайшего меньшего значения.

Если перед ключевым словом **цикл** указано слово **обратный**, то цикл выполняется в обратном порядке:

- Для строки — символы перебираются от конца строки к началу.
- Для массива — элементы перебираются от последнего к первому.
- Для словаря — ключи перебираются в порядке убывания.

Выполнение цикла можно прервать инструкцией **прервать**. Инструкция **продолжить** прерывает текущую итерацию цикла и немедленно начинает следующую — при том условии, что цикл ещё не дошёл до конца набора данных.

Примечание: модификация набора данных в теле цикла — добавление или удаление элементов массивов и словарей, вставка и удаление символов в строку — может приводить к непредсказуемым последствиям: к возникновению исключений или к неполному обходу набора данных.

Пока-цикл, цикл-пока — цикл с условием

В Клаусе есть две разновидности условного цикла: цикл с предусловием и цикл с постусловием.

Цикл с предусловием начинается ключевым словом **пока**, за которым следует выражение логического типа, слово **цикл**, тело цикла и знак «;».

Цикл с постусловием начинается ключевым словом **цикл**, за которым следует тело цикла, а после него слово **пока**, выражение логического типа и знак «;».

Телом цикла в обоих случаях должна быть ровно одна инструкция, в т. ч. это может быть составной блок, содержащий сколько угодно инструкций.

```
переменная
    д: дробное = 0;
начало
    пока д < 10 цикл д += 1.5;
    цикл д -= 1.2 пока д >= 0;
окончание.
```

Для цикла с предусловием значение логического выражения вычисляется перед первой итерацией. Если условие ложно, тело цикла не выполняется ни разу, а управление передаётся инструкции, следующей после цикла. Если условие истинно, тело цикла выполняется, после чего снова вычисляется значение логического выражения, и так происходит до тех пор, пока значение выражения не станет ложным.

Тело цикла с постусловием выполняется хотя бы один раз. После этого вычисляется значение логического выражения, и если оно истинно, тело цикла выполняется снова. Так происходит до тех пор, пока значение выражения не станет ложным, после чего управление передаётся инструкции, следующей после цикла.

Выполнение цикла можно прервать инструкцией **прервать**. Инструкция **продолжить** прерывает текущую итерацию цикла и немедленно переходит к следующей проверке условия, после которой цикл либо продолжается, либо завершается.

Прервать — прерывание цикла

Эта инструкция записывается ключевым словом **прервать**. Она прерывает выполнение цикла: инструкции в теле цикла, следующие после неё, не выполняются, и управление немедленно передаётся к ближайшей следующей инструкции после цикла.

Если в теле цикла есть составной блок, имеющий секцию **напоследок**, и в этом блоке была выполнена инструкция **прервать**, то инструкции в секции **напоследок** будут выполнены перед выходом из цикла.

Продолжить — прерывание итерации цикла

Эта инструкция записывается ключевым словом **продолжить**. Она прерывает выполнение текущей итерации цикла: инструкции в теле цикла, следующие после неё, не выполняются, и либо немедленно начинается следующая итерация, либо цикл завершается.

Если в теле цикла есть составной блок, имеющий секцию **напоследок**, и в этом блоке была выполнена инструкция **продолжить**, то инструкции в секции **напоследок** будут выполнены перед завершением итерации.

Стандартная библиотека

Базовый ввод-вывод

Особенности поведения функций ввода-вывода могут различаться в зависимости от реализации терминала, используемого для запуска программы.

Константы

```
константы
НС = ""№0Д№0А; // новая строка
Таб = №09; // символ табуляции
ВК = №0Д; // символ возврата каретки
ПС = №0А; // символ перевода строки
КФ = №1А; // символ конца файла
```

Процедура вывести()

процедура вывести(вх `arg0`, `arg1`, `arg2`, ...);

Выводит значения, переданные в параметрах, в стандартный поток вывода. По умолчанию поток вывода направлен на экран терминала, но он может быть переназначен средствами операционной системы при запуске программы.

Допускает любое количество параметров простых типов. Перед записью в поток значения параметров преобразуются в строковые прямым приведением типа.

При ошибках может вызывать исключение **ОшибкаВводаВывода**.

Примечание: для перевода строки нужно передать в параметре символ LF (**№0A**) или константу **НС**.

Процедура сообщить()

процедура сообщить(вх `arg0`, `arg1`, `arg2`, ...);

Выводит значения, переданные в параметрах, в стандартный поток сообщений об ошибках. По умолчанию этот поток направлен на экран терминала, но может быть переназначен средствами операционной системы при запуске программы.

Допускает любое количество параметров простых типов. Перед записью в поток значения параметров преобразуются в строковые прямым приведением типа.

При ошибках может вызывать исключение **ОшибкаВводаВывода**.

Примечание: для перевода строки нужно передать в параметре символ LF (**№0A**) или константу **НС**.

Функция ввести()

функция ввести(вых `arg0`, `arg1`, `arg2`, ...): целое;

Читает значения переданных параметров из стандартного потока ввода и возвращает количество прочитанных значений. По умолчанию поток ввода принимает пользовательский ввод с клавиатуры, но он может быть переназначен средствами операционной системы при запуске программы.

Допускает любое количество параметров простых типов. Считывает из потока все символы до ближайшего символа окончания строки (**№0A**, **№0D** или **№1A**), выбирает значения из полученной строки и по порядку присваивает их переданным параметрам. Выполняет следующие действия, в зависимости от типа переданного параметра:

- **Символ.** Выбирает следующий символ из прочитанной строки, конвертирует его в кодпойнт Unicode и присваивает параметру. Символ окончания строки считывается из потока, но не попадает в возвращаемое значение параметра.
- **Строка.** Присваивает параметру всю оставшуюся строку и возвращает управление. Символ окончания строки считывается из потока, но не попадает в возвращаемое значение параметра. Все параметры, следующие за строковым параметром, остаются непрочитанными — т. е., получают нулевые начальные значения.
- **Другие типы.** Пропускает пробелы и табуляции (**№20**, **№09**), считывает часть строки до ближайшего пробела или табуляции и приводит полученный текст к типу данных соответствующего по порядку параметра. Серия следующих друг за другом пробелов и табуляций считается одним разделителем.

Если в строке, полученной из потока, оказалось меньше данных, чем нужно для заполнения всех переданных параметров, то оставшиеся параметры не будут прочитаны и получают нулевые начальные значения. Если в строке больше данных, чем передано параметров, то оставшаяся часть строки будет отброшена.

Функция возвращает количество фактически прочитанных значений. При ошибках преобразования значений из их текстовых представлений создаётся исключение **ОшибкаКонвертации**, при ошибках чтения — исключение **ОшибкаВводаВывода**.

Примечание 1: чтение и запись значений моментов может быть взаимно неоднозначна, т. к. при приведении момента к строке текстовые представления даты и времени разделяются пробелом. Это может приводить к необходимости считывать момент в два параметра, расположенных друг за другом — сначала дату, затем время — и складывать полученные значения.

Примечание 2: не рекомендуется использовать эту функцию в сквозном режиме терминала, т. к. её поведение в этом режиме может быть неочевидным.

Функция прочестьСимвол()

функция прочестьСимвол() : символ;

Читает один символ из стандартного потока ввода и возвращает прочитанный символ.

В каноническом режиме терминала, если поток ввода пуст, функция ожидает, пока пользователь введёт с клавиатуры один или несколько символов и нажмёт **Enter**. После этого функция возвращает первый из введённых символов, а остальные символы остаются в потоке, так что следующие вызовы **прочестьСимвол()** выполняются немедленно. В том числе функция читает из потока и возвращает символ **№0A**, соответствующий нажатию клавиши **Enter**, а также любые другие управляющие символы (если терминал отправляет их в поток).

В сквозном режиме терминала, если поток ввода пуст, функция ожидает от пользователя нажатия любой клавиши и немедленно возвращает полученный символ. В этом режиме при нажатии **Enter** функция вернёт символ **№0D**.

При нажатии функциональных клавиш и клавиатурных комбинаций терминал может записывать в поток ввода управляющие последовательности, состоящие из нескольких символов. Для чтения таких последовательностей нужно вызвать функцию несколько раз подряд.

Если стандартный поток ввода перенаправлен на файл, эта функция последовательно возвращает символы из файла, а по достижении конца файла всегда возвращает символ **№1A**. При этом функция **естьСимвол()** всегда будет возвращать истинное значение.

Функция естьСимвол()

функция естьСимвол() : логическое;

Возвращает истинное значение, если в потоке ввода есть непрочитанные данные. В этом случае функция **прочестьСимвол()** не ожидает ввода от пользователя, а немедленно возвращает следующий символ из потока.

Если стандартный поток ввода перенаправлен на файл, эта функция всегда возвращает истинное значение.

Управление терминалом

Функции управления терминалом реализованы путём выдачи ESC-последовательностей в стандартный поток вывода или поток сообщений об ошибках, в соответствии со стандартом ECMA-48. Эти функции могут не работать или особенности их работы могут различаться в зависимости от реализации терминала, используемого для запуска программы.

Константы

```
константы
    // Идентификаторы стандартных потоков
    идСтдВывод = 1;
    идСтдСообщ = 2;
    // Режимы терминала
    трКанон = нет;
    трСквозной = да;
константы
    // Стили шрифта
    стшЖирный = 1; // жирный шрифт
    стшКурсив = 2; // курсив
    стшПодчерк = 4; // одинарное подчёркивание
    стшЗачерк = 8; // зачёркнутый текст
```

Процедура терминал()

```
процедура терминал(вх поток: целое; вх сквозной: логическое);
```

Выбирает один из двух стандартных потоков, с которым будут работать все последующие вызовы функций управления терминалом, и устанавливает для него режим работы.

Параметр **поток** может иметь одно из двух значений: **идСтдВывод** — стандартный поток вывода, **идСтдСообщ** — стандартный поток сообщений об ошибках.

Если параметр **сквозной** имеет значение **трСквозной**, то терминал переводится в режим сквозного ввода — все нажатия клавиш немедленно помещаются в поток ввода и отключается эхо-печать введённых символов.

Процедура размерЭкрана()

```
процедура размерЭкрана(вх горз, верт: целое);
```

Устанавливает размеры окна терминала. Параметр **горз** означает количество символов по горизонтали, а параметр **верт** — количество строк по вертикали.

Процедура очиститьЭкран()

```
процедура очиститьЭкран();
```

Очищает экран, используя установленный цвет шрифта и фона.

Процедура очиститьСтроку()

```
процедура очиститьСтроку(вх где: целое);
```

Очищает указанную часть строки в позиции курсора, используя установленный цвет шрифта и фона. Если переданное значение **где** отрицательно, очищает строку слева от курсора, включая позицию курсора. Если **где** положительно, очищает строку справа от курсора, включая позицию курсора. Нулевое значение очищает всю строку целиком.

Процедура курсор()

процедура курсор(вх горз, верт: целое);

Устанавливает курсор в переданную позицию. Нумерация строк на экране и символов в строке начинается с нуля.

Процедура курсорГорз()

процедура курсорГорз(вх горз: целое);

Устанавливает курсор в позицию переданного символа в текущей строке, вертикальная координата курсора не меняется. Нумерация символов в строке начинается с нуля.

Процедура курсорВерт()

процедура курсорВерт(вх верт: целое);

Устанавливает курсор в переданную строку на экране, горизонтальная координата курсора не меняется. Нумерация строк начинается с нуля.

Процедура подвинутьКурсор()

процедура подвинутьКурсор(вх горз, верт: целое);

Сдвигает курсор на указанное количество строк и/или столбцов относительно текущей позиции. Положительные значения параметров **горз** и **верт** сдвигают курсор вправо и вниз, а отрицательные — влево и вверх.

Процедура запомнитьКурсор()

процедура запомнитьКурсор();

Запоминает текущую позицию курсора. Для возврата курсора в сохранённую позицию используется процедура **вернутьКурсор()**.

Процедура вернутьКурсор()

процедура вернутьКурсор();

Возвращает курсор в позицию, которая была ранее сохранена вызовом процедуры **запомнитьКурсор()**.

Процедура скрытьКурсор()

процедура скрытьКурсор();

Скрывает мигающий прямоугольник курсора. Скрытый курсор продолжает выполнять свои функции как указатель текущей позиции для вывода текста.

Процедура показатьКурсор()

```
процедура показатьКурсор ();
```

Делает видимым мигающий прямоугольник курсора, если он был ранее скрыт.

Процедура цветФона()

```
процедура цветФона (вх цвет: целое);
```

Устанавливает переданный цвет фона, который будет использоваться при следующих вызовах функций вывода текста, а также очистки строк.

Параметр **цвет** может принимать значения от 0 до 255 и обозначает номер в 256-цветной палитре xTerm.

Процедура цветШрифта()

```
процедура цветШрифта (вх цвет: целое);
```

Устанавливает переданный цвет шрифта, который будет использоваться при следующих вызовах функций вывода текста.

Параметр **цвет** может принимать значения от 0 до 255 и обозначает номер в 256-цветной палитре xTerm.

Процедура стильШрифта()

```
процедура стильШрифта (вх стиль: целое);
```

Устанавливает переданный стиль шрифта, который будет использоваться при следующих вызовах функций вывода текста.

В параметре **стиль** можно передать нуль, либо одно или несколько значений констант **стшXXXX**, скомбинированных с помощью побитового ИЛИ. Например, выражение **стшЖирный | стшКурсив** означает жирный курсив, а выражение **стшЗачерк | стшПодчерк | стшКурсив** означает зачёркнутый и подчёркнутый курсив.

Процедура сброситьАтрибуты()

```
процедура сброситьАтрибуты ();
```

Сбрасывает ранее установленные цвет фона, цвет шрифта и стиль шрифта к значениям по умолчанию.

Функция цвет256()

```
функция цвет256 (вх кр, зел, син: целое): целое;
```

Возвращает ближайший подходящий цвет из 256-цветной палитры xTerm, наиболее точно соответствующий переданным значениям цветовых составляющих в модели RGB.

Параметры **кр**, **зел** и **син** соответствуют красной, зелёной и синей составляющим и могут принимать значения от 0 до 255.

Файловый ввод-вывод

Типы

```
тип
// Структура информации о файле, заполняемая
// функциями файлПервый() и файлСледующий()
ФайлИнфо = структура
    имя: строка; // имя файла
    размер: целое; // размер в байтах
    атрибуты: целое; // атрибуты, объединённые побитовым ИЛИ
    возраст: момент; // дата и время последней модификации
конец;
```

Константы

```
константы
// типы файлов
фтТекст = 0; // текстовый файл
фтДвоичный = 1; // файл с произвольным форматом данных
```

```
константы
// режимы открытия файла
фрдЧтение = $01; // разрешить чтение файла
фрдЗапись = $02; // разрешить запись в файл
```

```
константы
// режимы совместного доступа к файлу
фсдИсключить = $10; // закрыть другим процессам доступ к файлу
фсдЧтение = $20; // разрешить другим процессам чтение файла
фсдЛюбой = $40; // разрешить другим процессам любые действия с файлом
```

```
константы
// позиционирование в файле
фпОтНачала = 0; // считая от начала файла
фпОтКонца = 1; // считая от конца файла
фпОтносительно = 2; // считая от текущей позиции в файле
```

```
константы
// атрибуты файлов
фаТолькоЧтение = $00000001;
фаСкрытый = $00000002;
фаСистемный = $00000004;
фаМеткаТома = $00000008;
фаКаталог = $00000010;
фаАрхивный = $00000020;
фаНормальный = $00000080;
фаВременный = $00000100;
фаСимвСсылка = $00000400;
фаСжатый = $00000800;
фаШифрованный = $00004000;
фаВиртуальный = $00010000;
```

Функция файлСоздать()

функция файлСоздать(вх типФайла: целое; вх имя: строка; вх режим: целое): объект;

Создаёт новый файл с указанным именем в режиме чтения и записи. Если файл с таким именем уже существует, прежние данные из него будут удалены. При ошибках создаётся исключение **ОшибкаВводаВывода**.

Параметр **типФайла** означает тип данных создаваемого файла. Значение **фтТекст** означает текстовый файл в кодировке UTF-8, значение **фтДвоичный** — данные в произвольном формате. Функции **файлЗаписать ()** и **файлПрочсть ()** обращаются к файлу по-разному, в зависимости от значения этого параметра.

Параметр **режим** используется для указания режима совместного доступа к созданному файлу со стороны других процессов и может принимать одно из значений констант **фсдXXXX**.

Возвращаемое значение представляет собой объект, который нужно передавать первым параметром во все остальные функции файлового ввода-вывода.

Успешно созданный файл должен быть закрыт вызовом процедуры **файлЗакрыть ()**.

Функция файлОткрыть()

```
функция файлОткрыть (вх типФайла: целое; вх имя: строка; вх режим: целое): объект;
```

Открывает существующий файл с указанным именем и устанавливает позицию чтения-записи в начало файла. Если файл с таким именем не существует, а также при других ошибках создаётся исключение **ОшибкаВводаВывода**.

Параметр **типФайла** означает тип данных открываемого файла. Значение **фтТекст** означает текстовый файл в кодировке UTF-8, значение **фтДвоичный** — данные в произвольном формате. Функции **файлЗаписать ()** и **файлПрочсть ()** обращаются к файлу по-разному, в зависимости от значения этого параметра.

Параметр **режим** используется для указания режима доступа к файлу — только на чтение, только на запись, либо на чтение и запись, — а также режима совместного доступа к открытому файлу со стороны других процессов. Для этого используются константы **фрдXXXX** и **фсдXXXX**, которые можно комбинировать с помощью операции побитового «ИЛИ». Например, чтобы открыть файл для чтения и записи, запретив другим процессам запись в файл, но разрешив чтение, нужно передать в параметре **режим** выражение **фрдЧтение | фрдЗапись | фсдЧтение**.

Возвращаемое значение представляет собой объект, который нужно передавать первым параметром во все остальные функции файлового ввода-вывода.

Успешно открытый файл должен быть закрыт вызовом процедуры **файлЗакрыть ()**.

Процедура файлЗакрыть()

```
процедура файлЗакрыть (вв файл: объект);
```

Закрывает файл, открытый вызовом функции **файлСоздать ()** или **файлОткрыть ()**, и присваивает значение **пусто** переданному объекту.

Эта функция обязательно должна быть вызвана после успешного создания или открытия файла, иначе при завершении программы возникнет исключение **ОшибкаВыполнения**. Поэтому вызов этой функции желательно помещать в секцию **напоследок** составного блока, например:

```
переменная
  ф: объект;
начало
  ф := файлОткрыть(фтТекст, "мой-файл.txt", фрдЗапись | фсдИсключить);
начало
  файлРазмер(ф, 0);
```

```
    файлЗаписать (ф, "Вот такие числа:", НС);
    файлЗаписать (ф, 10, ' ', 20, ' ', 30, НС);
исключение
    когда любое тогда сообщить ("Ошибка при записи в файл.");
напоследок
    файлЗакреть (ф);
конец;
исключение
    когда любое тогда сообщить ("Ошибка при открытии файла.");
окончание.
```

Функция файлРазмер()

```
функция файлРазмер (вх файл: объект): целое;
функция файлРазмер (вх файл: объект; вх размер: целое): целое;
```

Возвращает размер переданного файла в байтах.

Если передан параметр **размер**, то устанавливает для файла переданный размер и возвращает новый размер файла. При этом позиция чтения-записи будет установлена в конец файла.

Функция файлПоз()

```
функция файлПоз (вх файл: объект): целое;
функция файлПоз (вх файл: объект; вх поз: целое): целое;
функция файлПоз (вх файл: объект; вх поз: целое; вх откуда: целое): целое;
```

Возвращает текущую позицию чтения-записи для переданного файла — номер байта, считая от начала файла. Первый байт в файле имеет номер 0.

Если передан параметр **поз**, то устанавливает новую позицию чтения-записи и возвращает установленное значение. Для записи в конец файла позиция должна указывать на байт, следующий за последним байтом в файле — т. е., быть равна размеру файла в байтах. Для текстовых файлов программист должен следить за тем, чтобы переданная позиция указывала на начало символа, в противном случае при попытке чтения возникнет исключение **ОшибкаВводаВывода**.

Параметр **откуда** определяет точку отсчёта для переданной позиции и может иметь одно из следующих значений:

- **фпОтНачала** — в **поз** передан номер байта, считая от начала файла. Это значение используется по умолчанию.
- **фпОтКонца** — в **поз** передан номер байта от конца файла. Значение 0 означает байт, следующий за концом файла, значение 1 — последний байт в файле, и т.п.
- **фпОтносительно** — в **поз** передано смещение относительно текущей позиции чтения-записи. Положительные значения перемещают позицию ближе к концу файла, отрицательные — к началу.

Процедура файлПрочеть()

```
процедура файлПрочеть (вх файл: объект; вых арг1, арг2, арг3, ...);
```

Читает данные из переданного файла, начиная с текущей позиции чтения-записи, и по порядку присваивает прочитанные значения переданным параметрам. Позиция чтения-записи сдвигается к концу файла на число прочитанных байтов. Если в файле недостаточно данных, чтобы прочесть все переданные параметры, возникает исключение **ОшибкаВводаВывода**.

Переданный файл должен быть открыт вызовом функции **файлОткрыть ()** или **файлСоздать ()**. Если файл был открыт как текстовый, то он должен содержать корректные символы в кодировке UTF-8, и текущая позиция чтения записи должна указывать на начало символа, иначе возникнет исключение **ОшибкаВводаВывода**.

Двоичный файл

При чтении данных из двоичного файла, в зависимости от типа каждого переданного параметра, процедура выполняет следующие действия:

- **строка** — читает из файла 64-битное целое со знаком, представляющее собой длину строки в байтах. Затем, если прочитанная длина больше нуля, читает соответствующее число байтов и заполняет ими содержимое строки. При этом корректность прочитанных символов UTF-8 не проверяется, что может приводить к различным ошибкам при дальнейших обращениях к прочитанной строке.
- **объект** — значение этого типа не может быть прочитано из файла. При попытке чтения возникнет исключение **НеверныйТип**.
- **другие простые типы** — читает из файла один или несколько байтов, в соответствии с размером внутреннего представления значения данного типа, как это описано в разделе **Типы данных**.
- **составные типы** — значения составных типов не могут быть прочитаны из файла.

Текстовый файл

При чтении данных из текстового файла, в зависимости от типа каждого переданного параметра, процедура выполняет следующие действия:

- **символ** — читает из файла один символ в кодировке UTF-8 и приводит его к 32-битному целому, означающему кодпойнт Юникод.
- **строка** — читает из файла все символы до ближайшего символа окончания строки (**№ОД**, **№ОА** или их комбинация), либо до конца файла. Символы окончания строки будут прочитаны из файла, но не попадут в возвращаемое строковое значение. Если текущая позиция чтения-записи указывает на символ окончания строки, прочитанное строковое значение будет пустым.
- **объект** — значение этого типа не может быть прочитано из файла. При попытке чтения возникнет исключение **НеверныйТип**.
- **другие простые типы** — читает из файла все символы до ближайшего символа-разделителя: пробела, табуляции, возврата каретки, перевода строки, либо до конца файла. Если текущая позиция чтения-записи указывает на символ-разделитель, то он и следующие за ним разделители будут пропущены. Полученное строковое значение будет приведено к соответствующему типу данных; если это невозможно, возникнет исключение **ОшибкаКонвертации**.
- **составные типы** — значения составных типов не могут быть прочитаны из файла.

Процедура файлЗаписать()

```
процедура файлЗаписать(вх файл: объект; вх arg1, arg2, arg3, ...);
```

Записывает в файл значения переданных параметров, начиная с текущей позиции чтения-записи. Если текущая позиция чтения-записи меньше размера файла в байтах, то новые данные будут записаны поверх старых; в противном случае размер файла будет увеличен до необходимого. Текущая позиция чтения-записи смещается к концу файла на число записанных байтов. При ошибках записи возникает исключение **ОшибкаВводаВывода**.

Переданный файл должен быть открыт вызовом функции **файлОткрыть ()** или **файлСоздать ()**.

Двоичный файл

При записи данных в двоичный файл, в зависимости от типа каждого переданного параметра, процедура выполняет следующие действия:

- **строка** — записывает в файл 64-битное целое со знаком, представляющее собой длину строки в байтах. Затем, если строка непустая, записывает один или несколько байтов — содержимое строки в кодировке UTF-8.
- **другие простые типы** — записывает в файл один или несколько байтов, в соответствии с размером внутреннего представления значения данного типа, как это описано в разделе **Типы данных**.
- **составные типы** — значения составных типов не могут быть записаны в файл.

Текстовый файл

При записи данных в текстовый файл, в зависимости от типа каждого переданного параметра, процедура выполняет следующие действия:

- **символ** — записывает в файл один символ в кодировке UTF-8.
- **строка** — записывает в файл символы в кодировке UTF-8, составляющие содержимое строки. Никакие дополнительные символы, свидетельствующие об окончании строки, в файл не записываются.
- **другие простые типы** — записывает в файл строковое представление переданного значения, полученное прямым приведением типа. Никакие дополнительные символы-разделители в файл не записываются.
- **составные типы** — значения составных типов не могут быть записаны в файл.

Функция файлЕсть()

функция файлЕсть(вх имя: строка): логическое;

Возвращает истинное значение, если существует файл, но не каталог, с указанным именем.

Функция файлЕстьКат()

функция файлЕстьКат(вх имя: строка): логическое;

Возвращает истинное значение, если существует каталог с указанным именем.

Функция файлВрмКат()

функция файлВрмКат(вх общий: логическое): строка;

Возвращает путь к каталогу для размещения временных файлов. Если в параметре **общий** передано истинное значение, функция вернёт глобальный временный каталог, в противном случае — временный каталог текущего пользователя, если это поддерживается операционной системой.

Функция файлВрмИмя()

функция файлВрмИмя(вх общий: логическое; вх префикс: строка): строка;

Возвращает путь и имя файла, которые можно использовать для создания временного файла. Если в параметре **общий** передано истинное значение, функция вернёт путь к файлу в глобальном временном каталоге, в

противном случае — во временном каталоге текущего пользователя, если это поддерживается операционной системой.

Имя файла состоит из переданного префикса и уникального суффикса из нескольких цифр. Если в параметре **префикс** передана пустая строка, будет использован префикс **"TMP"**.

Функция файлПолныйПуть()

функция файлПолныйПуть(вх путь: строка): строка;

Дополняет переданный относительный путь к файлу или каталогу до абсолютного, с учётом текущего каталога выполнения программы, и заменяет символы-разделители пути на принятые в данной операционной системе.

Если в параметре **путь** передано пустое значение, возвращает полный путь к текущему каталогу.

Функция файлВыполняемый()

функция файлВыполняемый(): строка;

Возвращает имя файла выполняемой программы.

Функция файлДомКат()

функция файлДомКат(): строка;

Возвращает путь к домашнему каталогу текущего пользователя операционной системы.

Функция файлАтрибуты()

функция файлАтрибуты(вх имя: строка): целое;

Возвращает атрибуты файла — целочисленное значение, представляющее собой комбинацию всех установленных атрибутов файла, объединённых операцией побитового ИЛИ. При ошибках создаёт исключение **ОшибкаВводаВывода**.

Возможным значениям атрибутов соответствуют константы **фаXXXX**. Некоторые атрибуты могут быть неприменимы, в зависимости от операционной системы.

Функция файлВозраст()

функция файлВозраст(вх имя: строка): момент;

Возвращает дату и время последней модификации файла. При ошибках создаёт исключение **ОшибкаВводаВывода**.

Функция файлПереместить()

процедура файлПереместить(вх откуда, куда: строка);

Переименовывает и/или перемещает файл с переданными именами. При ошибках создаёт исключение **ОшибкаВводаВывода**.

Функция файлУдалить()

процедура файлУдалить (вх имя: строка);

Удаляет файл с переданными именем. При ошибках создаёт исключение **ОшибкаВводаВывода**.

Функция файлПервый()

функция файлПервый (вх шаблон: строка; вых инфо: ФайлИнфо): объект;

Создаёт объект — список файлов, подходящих под переданный **шаблон**, и заполняет переданную структуру данными о первом из найденных файлов.

Шаблон может содержать абсолютный или относительный путь к каталогу, а также маску имени файла с символами «*» и «?».

Если функция нашла один или несколько подходящих файлов, то она возвращает непустой объект, который можно передавать в функцию **файлСледующий()** для получения информации о следующих найденных файлах. Этот объект должен быть в дальнейшем уничтожен вызовом процедуры **уничтожить()**. Если подходящие файлы не найдены, функция вернёт значение **пусто**.

Пример

```
// Выводит список файлов и каталогов в домашнем каталоге пользователя,  
// за исключением скрытых файлов и каталогов.  
программа СписокФайлов;  
переменные  
    шаблон: строка;  
    поиск: объект;  
    инфо: ФайлИнфо;  
    найдено: логическое;  
начало  
    шаблон := файлПолныйПуть (файлДомКат() ++ "*");  
    вывести(шаблон, НС);  
    поиск := файлПервый(шаблон, инфо);  
    если поиск <> пусто то начало  
        цикл начало  
            если инфо.атрибуты & фаСкрытый = 0 то начало  
                если инфо.атрибуты & фаКаталог <> 0 то  
                    вывести(формат (  
                        "%20с %8ш %19м %с",  
                        "<каталог>", инфо.атрибуты, инфо.возраст, инфо.имя), НС)  
                иначе  
                    вывести(формат (  
                        "%20ц %8ш %19м %с",  
                        инфо.размер, инфо.атрибуты, инфо.возраст, инфо.имя), НС);  
            конец;  
            найдено := файлСледующий(поиск, инфо);  
        конец пока найдено;  
    напоследок  
        уничтожить (поиск);  
конец;  
окончание.
```

Функция файлСледующий()

```
функция файлСледующий(вх список: объект; вых инфо: ФайлИнфо): логическое;
```

Заполняет переданную структуру данными о следующем найденном файле из переданного списка и возвращает истинное значение. Если в списке больше нет файлов, возвращает значение **нет**; в этом случае структура **инфо** остаётся незаполненной.

Объект **список** должен быть создан вызовом функции **файлПервый()** и в дальнейшем уничтожен вызовом процедуры **уничтожить()**.

Пример использования см. в описании функции **файлПервый()**.

Массивы

Конструкции есть() и нету()

```
переменная
  м: массив целых;
начало
  если есть(м[10]) то вывести("да");
  если нету(м[10]) то вывести("нет"); // будет выведено "нет"
окончание;
```

Эти конструкции выглядят как функции, но фактически ими не являются. Конструкция **есть()** имеет истинное значение, если существует указанный в скобках элемент массива — т. е., если указанный индекс принадлежит диапазону от 0 до длины массива минус 1. Конструкция **нету()** имеет истинное значение в противоположном случае.

Функция длина()

```
функция длина(вх м: массив): целое;
функция длина(вв м: массив; вх размер: целое): целое;
```

Возвращает количество элементов в переданном массиве.

Если передан параметр **размер**, то функция установит для переданного массива указанную длину и вернёт новое значение длины. В этом случае в первом параметре должна быть передана переменная.

Если новая длина больше старой, то прежнее содержимое массива сохраняется, а новые элементы добавляются в конец массива и инициализируются нулевыми начальными значениями. Если новая длина меньше старой, то сохраняются прежние значения элементов от нуля до новой длины, а остальные элементы уничтожаются.

Процедура добавить()

```
процедура добавить(вв м: массив; вх элемент);
```

Добавляет переданный элемент в конец массива. Длина массива увеличивается на 1.

Процедура вставить()

```
процедура вставить(вв м: массив; вх индекс: целое; вх элемент);
```

Вставляет в массив переданный элемент. Если переданный **индекс** равен длине массива, то элемент добавляется в конец массива; если меньше, то элемент вставляется в переданную позицию, а все элементы, начиная от переданного индекса, включительно, сдвигаются в сторону увеличения индекса. Длина массива увеличивается на 1.

Если передан недопустимый индекс, создаётся исключение **НеверныйИндекс**.

Процедура удалить()

```
процедура удалить (вв м: массив; вх индекс: целое);  
процедура удалить (вв м: массив; вх индекс, кво: целое);
```

Удаляет элементы из массива.

Удаляет переданное количество элементов, начиная от указанного индекса, включительно. Если параметр **кво** не передан, удаляет один элемент. Если после удалённых элементов остались ещё элементы, то они сдвигаются в сторону уменьшения индекса.

Если передан недопустимый индекс или слишком большое количество, создаётся исключение **НеверныйИндекс**.

Процедура очистить()

```
процедура очистить (вв м: массив);
```

Удаляет все элементы из массива.

Словари

Конструкции есть() и нету()

```
переменная  
  сл: словарь целых ключ строка;  
начало  
  сл["первый"] := 10;  
  если есть (сл["первый"]) то вывести("да"); // будет выведено "да"  
  если нету (сл["первый"]) то вывести("нет");  
окончание;
```

Эти конструкции выглядят как функции, но фактически ими не являются. Конструкция **есть()** имеет истинное значение, если существует указанный в скобках элемент словаря — т. е., если в словаре есть указанный ключ. Конструкция **нету()** имеет истинное значение в противоположном случае.

Функция длина()

```
функция длина (вх сл: словарь): целое;
```

Возвращает количество элементов в переданном словаре.

Процедура добавить()

```
процедура добавить (вв сл: словарь; вх ключ);  
процедура добавить (вв сл: словарь; вх ключ, значение);
```

Добавляет в словарь переданный **ключ**. Если передано **значение**, то оно присваивается добавленному ключу; если нет, то новый ключ получает нулевое начальное значение.

Если указанный ключ уже существует в словаре, то новое значение присваивается существующему ключу, либо не меняется, если новое значение не передано.

Процедура вставить()

```
процедура вставить (вв сл: словарь; вх ключ, значение);
```

Добавляет в словарь переданную пару ключ-значение или присваивает новое значение переданному ключу, если тот уже существует в словаре. Эквивалентным вызову этой функции является присваивание:

```
сл[ключ] := значение;
```

Процедура удалить()

```
процедура удалить (вв сл: словарь; вх ключ);
```

Удаляет из словаря переданный ключ и соответствующее ему значение. Если указанный ключ отсутствует в словаре, создаётся исключение **НеверныйКлюч**.

Процедура очистить()

```
процедура очистить (вв сл: словарь);
```

Удаляет из словаря все ключи и соответствующие им значения.

Числа

Константы

```
константы
МинЦелое = -9223372036854775808; // Минимальное значение целого числа
максЦелое = 9223372036854775807; // Максимальное значение целого числа
минДробное = 2.2250738585072E-308; // Минимальное значение дробного числа
максДробное = 1.79769313486232E308; // Максимальное значение дробного числа
Pi = 3.14159265358979; // Значение числа Пи
Пи = 3.14159265358979; // Значение числа Пи
```

Функция нечисло()

```
функция нечисло (вх число: дробное): логическое;
```

Возвращает истину, если переданный параметр имеет значение **нечисло**.

Функция конечно()

```
функция конечно (вх число: дробное): логическое;
```

Возвращает истину, если переданный параметр имеет конечное численное значение. Возвращает ложь, если значение переданного параметра — **нечисло** или бесконечность с любым знаком.

Функция округл()

функция округл(вх число: дробное) : целое;

Округляет переданное значение до ближайшего целого и возвращает полученное целое число. Если передано **нечисло** или бесконечность, возникает исключение **НеверноеЧисло**.

Функция цел()

функция цел(вх число: дробное) : дробное;

Возвращает целую часть переданного числа.

Функция дроб()

функция дроб(вх число: дробное) : дробное;

Возвращает дробную часть переданного числа.

Функция sin()

функция sin(число: дробное) : дробное;

Возвращает синус переданного числа.

Функция cos()

функция cos(число: дробное) : дробное;

Возвращает косинус переданного числа.

Функция tg()

функция tg(число: дробное) : дробное;

Возвращает тангенс переданного числа.

Функция arcsin()

функция arcsin(число: дробное) : дробное;

Возвращает арксинус переданного числа.

Функция arccos()

функция arccos(число: дробное) : дробное;

Возвращает арккосинус переданного числа.

Функция arctg()

функция arctg(число: дробное) : дробное;

Возвращает арктангенс переданного числа.

Функция ln()

функция ln(число: дробное) : дробное;

Возвращает натуральный логарифм переданного числа.

Функция exp()

функция exp(число: дробное) : дробное;

Возвращает экспоненту переданного числа — т. е., число, возведённое в степень e .

Функция случайное()

функция случайное(макс: целое) : целое;

Возвращает случайное число в диапазоне от 0 включительно до переданного значения **макс**, но не включая **макс**. Генератор псевдослучайных чисел инициализируется при первом обращении к функции.

Моменты

Функция дата()

функция дата() : момент;
функция дата(м: момент) : момент;

Возвращает значение переданного момента с обнулённым временем — т. е., только дату. Если параметр не передан, возвращает текущую дату.

Функция время()

функция время() : момент;
функция время(м: момент) : момент;

Возвращает значение переданного момента с обнулённой датой — т. е., только время. Если параметр не передан, возвращает текущее время.

Функция сейчас()

функция сейчас() : момент;

Возвращает значение текущего момента — дату и время.

Исключения

Предопределённые исключения

В стандартной библиотеке Клаус определены следующие исключения:

- ОшибкаВводаВывода** - возникает при ошибках чтения и записи данных в стандартные потоки ввода-вывода или файлы, при неуспешном завершении функций управления файлами и в других подобных случаях.
- ОшибкаКонвертации** - возникает при ошибках преобразования значений из текстового формата в двоичный и обратно.
- НеверныйИндекс** - возникает при обращении к несуществующему индексу в массиве.
- НеверныйКлюч** - возникает при обращении к несуществующему ключу в словаре.
- НеверныйСимвол** - возникает при ошибках, связанных с обращением к неверному байту в строке.
- НеверноеИмя** - возникает при обращении к несуществующему идентификатору.
- НеверноеЧисло** - возникает при ошибках, связанных с обработкой дробных чисел.
- НеверныйТип** - возникает при ошибках, связанных с несовместимостью типов данных.
- НеверныйСинтаксис** - возникает при синтаксических ошибках.
- ОшибкаВыполнения** - прочие ошибки программы или операционной системы.
- ВнутренняяОшибка** - ошибки, допущенные разработчиками Клауса. При возникновении такой ошибки авторы будут благодарны за добавление сообщения о проблеме в репозиторий проекта. Заранее спасибо!

Все предопределённые исключения не принимают параметров. Программист может создать любое из перечисленных исключений по своему усмотрению.

Функция имя_исключения()

```
функция имя_исключения() : строка;
```

Возвращает имя активного исключения. Эту функцию можно вызывать только в секции обработки исключений, иначе она вернёт пустую строку.

Функция текст_исключения()

```
функция текст_исключения() : строка;
```

Возвращает текст сообщения об ошибке активного исключения с уже добавленными в него значениями параметров исключения. Эту функцию можно вызывать только в секции обработки исключений, иначе она вернёт пустую строку.

Строки

Во всех строковых функциях стандартной библиотеки Клаус, если иное не оговорено явно, целочисленные значения индексов и размеров имеют семантику байтов, а не символов — как если бы строка представляла собой не последовательность символов переменной длины, а массив однобайтовых целых чисел.

Байты в строке нумеруются начиная с нуля: первый байт имеет индекс 0, а последний имеет индекс на единицу меньший, чем длина строки.

Библиотечные функции поддерживают внутреннюю целостность строки, не допуская появления в ней некорректных символов, и вызывают исключение **НеверныйСимвол** в случае ошибок. Например, такое исключение возникнет при вставке подстроки в позицию, соответствующую середине многобайтового символа, при удалении из строки части многобайтового символа, а также в других аналогичных ситуациях.

Функция длина()

```
функция длина (вх стр: строка) : целое;  
функция длина (вв стр: строка; вх размер: целое) : целое;
```

Возвращает длину переданной строки в байтах. Следует иметь в виду, что количество байтов в строке не равняется количеству символов, т. к. в кодировке UTF-8 символ может занимать от 1 до 4 байтов.

Если передан параметр **размер**, то функция установит для переданной строки указанную длину в байтах и вернёт новое значение длины. В этом случае в первом параметре должна быть передана строковая переменная.

Если новая длина больше старой, то прежнее содержимое строки сохраняется, и строка дополняется справа пробелами (**№20**) до указанной длины.

Если новая длина меньше старой, то прежнее содержимое строки обрезается справа до указанной длины. В этом случае должна быть указана такая новая длина, чтобы последний символ в строке, если он занимает несколько байтов, не оказался разбит на части — иначе возникнет ошибка **НеверныйСимвол**.

Пример

```
переменная  
    стр: строка;  
начало  
    // Буквы кириллицы занимают в кодировке UTF-8  
    // по два байта, а символ пробела - один байт.  
    стр := "АВВГ";  
    длина (стр, 10); // стр = "АВВГ "  
    длина (стр, 6); // стр = "АВВ"  
    длина (стр, 5); // ошибка  
окончание.
```

Функция симв()

```
функция симв (вх стр: строка) : символ;  
функция симв (вх стр: строка; вх индекс: целое) : символ;
```

Возвращает символ, расположенный в переданной позиции строки **стр**. Строка не должна быть пустой; переданный индекс должен быть в пределах строки и должен соответствовать началу символа в строке, иначе возникнет исключение **НеверныйСимвол**.

Если параметр **индекс** не указан, он считается равным 0, и функция возвращает первый символ в строке.

Функция след()

```
функция след (вх стр: строка; вх индекс: целое) : целое;  
функция след (вх стр: строка; вх индекс, кво: целое) : целое;  
функция след (вх стр: строка; вх индекс, кво: целое; вых симв: целое) : целое;
```

Возвращает позицию в байтах, соответствующую началу следующего символа в строке, считая от символа, переданного в параметре **индекс**. Значение переданного индекса должно соответствовать началу символа в строке, иначе возникнет исключение **НеверныйСимвол**.

В параметре **кво** можно передать количество символов, на которое необходимо сместиться: 1 — следующий символ, 2 — через один символ, и т. п. Если переданное количество превышает число оставшихся (справа от **индекс**) символов в строке, то функция вернёт длину строки в байтах.

Если передан параметр **симв**, то функция запишет в него фактически пройденное количество символов. Таким образом, чтобы посчитать общее количество символов в строке, можно передать 0 в параметре **индекс** и **максЦелое** в параметре **кво**.

Функция пред()

```
функция пред(вх стр: строка; вх индекс: целое): целое;  
функция пред(вх стр: строка; вх индекс, кво: целое): целое;  
функция пред(вх стр: строка; вх индекс, кво: целое; вых симв: целое): целое;
```

Возвращает позицию в байтах, соответствующую началу предыдущего символа в строке, считая от символа, переданного в параметре **индекс**. Для получения позиции последнего символа в строке в параметре **индекс** нужно передать значение, большее или равное длине строки в байтах. Если переданный индекс находится в пределах строки, то он должен соответствовать началу символа в строке, иначе возникнет исключение **НеверныйСимвол**.

В параметре **кво** можно передать количество символов, на которое необходимо сместиться: 1 — предыдущий символ, 2 — через один символ, и т. п. Если переданное количество превышает число оставшихся (слева от **индекс**) символов до начала строки, то функция вернёт 0.

Если передан параметр **симв**, то функция запишет в него фактически пройденное количество символов. Таким образом, чтобы посчитать общее количество символов в строке, можно передать **максЦелое** в параметрах **индекс** и **кво**.

Примечание: для строки из многобайтовых символов функция **след()** работает быстрее, чем **пред()**.

Функция часть()

```
функция часть(вх стр: строка; вх индекс: целое): строка;  
функция часть(вх стр: строка; вх индекс, размер: целое): строка;
```

Возвращает часть строки указанного размера, начиная с указанного индекса. Значение переданного индекса должно соответствовать началу символа в строке, иначе возникнет исключение **НеверныйСимвол**.

Если параметр **размер** не передан или выходит за пределы строки, то функция вернёт часть строки от указанного индекса до конца строки. В противном случае, **размер** должен иметь такое значение, чтобы последний символ в возвращаемой подстроке, если он занимает несколько байтов, не оказался разбит на части — иначе возникнет исключение **НеверныйСимвол**.

Процедура добавить()

```
процедура добавить(вв стр1: строка; вх стр2: строка);
```

Добавляет значение **стр2** в конец строки **стр1**. Эквивалентным вызову этой функции является присваивание:

```
стр1 := стр1 ++ стр2;
```

Процедура вставить()

процедура вставить (вв *стр*: строка; вх *индекс*: целое; вх *подстр*: строка);

Вставляет строку **подстр** в указанную позицию строки **стр**, так что символы из строки **подстр** окажутся в строке **стр** непосредственно перед тем символом, на который указывает **индекс**. При этом длина строки **стр** увеличивается на длину строки **подстр**.

Если переданный **индекс** больше или равен длине **стр**, то подстрока будет добавлена к строке справа. В противном случае значение индекса должно соответствовать началу символа в строке, иначе возникнет исключение **НеверныйСимвол**.

Процедура удалить()

процедура удалить (вв *стр*: строка; вх *индекс*, *число*: целое);

Удаляет из строки указанное число байтов, начиная с переданного индекса, включительно.

Значение индекса должно соответствовать началу символа в строке, иначе возникнет исключение **НеверныйСимвол**.

Если строка содержит меньше байтов справа от индекса, чем передано в параметре **число**, то будут удалены все байты до конца строки. В противном случае, переданное число должно быть таким, чтобы последний удаляемый символ, если он занимает несколько байтов, не оказался разбит на части — иначе возникнет исключение **НеверныйСимвол**.

Процедура вписать()

процедура вписать (
вв *стр1*: строка; вх *индекс*: целое; вх *стр2*: строка);

процедура вписать (
вв *стр1*: строка; вх *индекс*: целое;
вх *стр2*: строка; вх *индекс2*: целое);

процедура вписать (
вв *стр1*: строка; вх *индекс*: целое;
вх *стр2*: строка; вх *индекс2*, *число*: целое);

Копирует в **стр1** байты из **стр2**.

Из **стр2** копируется указанное **число** байтов, начиная от **индекс2**, включительно; эти байты размещаются в **стр1**, начиная от **индекс**, и заменяют собой байты, которые раньше находились в **стр1** в этих позициях. Если длины **стр1** не хватает, чтобы разместить все копируемые байты, то её длина увеличивается до необходимой.

Если не передано **число**, то копируются все байты от **индекс2** до конца **стр2**. Если не передан **индекс2**, то копируется вся **стр2** от начала до конца.

Все индексы и размеры должны иметь такие значения, чтобы никакой многобайтовый символ в обеих строках не оказался при копировании разбит на части, иначе возникнет ошибка **НеверныйСимвол**.

Пример

```
переменная
    стр1, стр2: строка;
начало
    // Буквы кириллицы занимают в кодировке UTF-8
    // по два байта, а арабские цифры - один байт.
```

```

стр1 := "1234АБВГ";
стр2 := "0000";
вписать(стр1, 2, стр2);           // стр1 = 120000БВГ
вписать(стр1, максЦелое, стр2, 0, 2); // стр1 = 120000БВГ00
вписать(стр1, 7, стр2);           // ошибка
вписать(стр1, 6, стр2, 0, 3);     // ошибка
окончание.

```

Функция найти()

```

функция найти(
    вх что, где: строка): целое;
функция найти(
    вх что, где: строка; вх начИдкс: целое): целое;
функция найти(
    вх что, где: строка; вх начИдкс: целое;
    вх учтРегистр: логическое): целое;

```

Ищет подстроку **что** в строке **где** и возвращает индекс первого байта найденной подстроки. Если подстрока не найдена, возвращает -1. Если передан **начИдкс**, поиск начинается с переданного индекса, включительно.

Если в параметре **учтРегистр** передано **нет**, то регистр символов при поиске не учитывается — т. е., прописные буквы считаются равными строчным. Если этот параметр опущен, функция выполняет поиск с учётом регистра.

Процедура заменить()

```

процедура заменить(вв стр: строка; вх индекс, число: целое; вх чем: строка);

```

Заменяет в переданной строке указанное **число** байтов, начиная с переданного индекса, на байты из строки **чем**. Длина строки изменяется соответственно.

Значения **индекс** и **число** должны быть такими, чтобы в результате замены никакой многобайтовый символ не оказался разбит на части, иначе возникнет ошибка **НеверныйСимвол**.

Функция формат()

```

функция формат(вх шаблон: строка; вх арг0, арг1, арг2, ...): строка;

```

Заменяет все спецификаторы в переданном шаблоне на значения переданных аргументов, преобразованные к строке в соответствии с указанным форматом, и возвращает полученную строку.

Спецификаторы имеют следующий синтаксис:

```
'%' [индекс ':' ] [ '-' ] [ширина] [ '.' точность ] Тип
```

Символы в апострофах должны быть указаны в шаблоне буквально, без апострофов.

Элементы в квадратных скобках необязательны.

'%'	Начало спецификатора. Чтобы добавить в шаблон просто символ '%', его нужно удвоить: "%%".
индекс ':'	Целое. Взять из списка параметров аргумент с указанным индексом. Если этот элемент не указан, берётся следующий по порядку аргумент.

' - '	Выравнивание вставляемого текста по левому краю поля. По умолчанию текст выравнивается по правому краю. Этот элемент работает только при условии, что указана ширина.
ширина	Целое. Ширина поля — т. е., минимальная длина вставляемой строки. Если строка короче, то она дополняется до указанной длины пробелами слева (по умолчанию) или справа, если указан «-».
'.'точность	Целое. Значение этого элемента зависит от типа спецификатора.
Тип	Тип спецификатора — буква в любом регистре. Перечень допустимых типов приведён ниже.

В случае ошибок функция создаёт исключение **ОшибкаКонвертации**. Это может происходить при ошибках в синтаксисе спецификаторов, при несоответствии типов спецификаторов типам аргументов, а также если передано недостаточно аргументов для всех спецификаторов.

Типы спецификаторов

«I» или «Ц» — целое число. Аргумент должен быть целым. Значение конвертируется в строковое представление целого числа. Если указана точность, то строка будет содержать не менее указанного количества цифр — если нужно, слева будут добавлены нули.

«E» (лат.) или «Э» — экспоненциальный формат. Аргумент может быть целым или дробным. Число приводится к строке в экспоненциальном (научном) формате. Точность определяет общее количество цифр в мантиссе числа, порядок записывается тремя цифрами.

«F» или «Ф» — формат с фиксированной точкой. Аргумент может быть целым или дробным. Точность определяет количество цифр после десятичной точки. Для слишком больших чисел используется экспоненциальный формат.

«G» или «O» (кир.) — общий числовой формат. Аргумент может быть целым или дробным. Число приводится к строке в фиксированном или экспоненциальном формате в зависимости от того, что получится короче. Точность определяет общее количество цифр в строковом представлении числа.

«N» или «Ч» — такой же формат числа, как с фиксированной точкой, но с добавлением разделителей групп разрядов.

«D» или «Д» — дата в стандартном формате: YYYY-MM-DD. Аргумент должен быть моментом. Точность не применяется.

«T» (лат.) или «В» (кир.) — время в стандартном формате: hh:mm:ss. Аргумент должен быть моментом. Точность не применяется.

«M» (лат.) или «М» (кир.) — дата и время в стандартном формате, разделённые пробелом. Аргумент должен быть моментом. Точность не применяется.

«S» или «С» (кир.) — строка. Аргумент может быть любого простого типа — строковые значения копируются, а значения других типов конвертируются в строку прямым приведением типа. Точность не применяется.

«X» (лат.) или «Ш» — шестнадцатеричное представление целого числа. Аргумент может быть целым, символом или объектом. Точность определяет минимальное количество цифр — если нужно, слева будут добавлены нули.

Пример

```
стр := формат("Меня зовут %s %s. Мне %i лет.", "Пётр", "Иванов", 15);
// стр = "Меня зовут Пётр Иванов. Мне 15 лет."
```

```

стр := формат("Меня зовут %1:s %0:s. Мне %2:i лет.", "Пётр", "Иванов", 15);
// стр = "Меня зовут Иванов Пётр. Мне 15 лет."

стр := формат("[%i]", 10); // стр = "[10]"
стр := формат("[%%]", 10); // стр = "[%]"
стр := формат("[%10i]", 10); // стр = "[ 10]"
стр := формат("[%4i]", -10); // стр = "[-0010]"
стр := формат("[%10.4ц]", -10); // стр = "[ -0010]"
стр := формат("[%0:-10i]", 10); // стр = "[10 ]"
стр := формат("[%x]", 1234567); // стр = "[12D687]"
стр := формат("[%x]", -1234567); // стр = "[FFFFFFFFFED2979]"
стр := формат("[%0:10x]", 'Ё'); // стр = "[ 401]"
стр := формат("[%0:10.4ш]", 'Ё'); // стр = "[ 0401]"
стр := формат("[%12.4e]", -1.234); // стр = "[ -1.234E+000]"
стр := формат("[%12.4f]", -1.234); // стр = "[ -1.2340]"
стр := формат("[%g]", 1.234); // стр = "[1.234]"
стр := формат("[%12g]", -1.234); // стр = "[ -1.234]"
стр := формат("[%0:-12.4G]", 1.234); // стр = "[1.234 ]"
стр := формат("[%14d]", момент(44567.123)); // стр = "[ 2022-01-06]"
стр := формат("[%14д]", момент(44567.123)); // стр = "[2022-01-06 ]"
стр := формат("[%14t]", момент(44567.123)); // стр = "[ 02:57:07]"
стр := формат("[%14в]", момент(44567.123)); // стр = "[02:57:07 ]"
стр := формат("[%s]", "Привет!"); // стр = "[Привет!]"
стр := формат("[%14с]", "Привет!"); // стр = "[ Привет!]"
стр := формат("[%14с]", "Привет!"); // стр = "[Привет! ]"

```

Функция загл()

```
функция загл(вх стр: строка): строка;
```

Возвращает строку, в которой все строчные буквы заменены на прописные.

Функция строч()

```
функция строч(вх стр: строка): строка;
```

Возвращает строку, в которой все прописные буквы заменены на строчные.

Графический вывод

Все функции графического вывода стандартной библиотеки Клаус работают только в том случае, если программа запущена в отладочной среде. При запуске программы из системного терминала попытка создания графического окна, а также обращение к любым другим функциям графического вывода, вызывает исключение **ОшибкаВыполнения**.

Все функции графического вывода используют систему координат, где точкой начала отсчёта считается левый верхний угол графического окна.

Типы

```
тип
Точка/Точки/Точек = структура
г, в: целое;
```

окончание;

МассивТочек/МассивыТочек/МассивовТочек = массив Точек;

Константы

```
константы
// стили пера
грспПусто      = 0;
грспЛиния     = 1;
грспТочка     = 2;
грспТире      = 3;
грспТочкаТире = 4;
грсп2ТочкиТире = 5;
константы
// стили кисти
грскПусто = 0;
грскЦвет  = 1;
константы
// стили шрифта
грсшЖирный  = 1; // жирный шрифт
грсшКурсив  = 2; // курсив
грсшПодчерк = 4; // одинарное подчёркивание
грсшЗачерк  = 8; // зачёркнутый текст
```

Функция грОткрытьОкно()

функция грОткрытьОкно(вх заголовок: строка): объект;

Создаёт новое графическое окно — закладку в окне отладочной консоли, предназначенную для вывода данных в графическом режиме. Название новой закладки должно быть передано в параметре **заголовок**.

Возвращает объект, соответствующий созданному окну — этот объект нужно передавать первым параметром во все остальные функции графического вывода. Перед завершением работы программы все графические окна должны быть закрыты вызовом процедуры **грЗакретьОкно()**.

Процедура грЗакретьОкно()

процедура грЗакретьОкно(вв окно: объект);

Закрывает графическое окно, открытое функцией **грОткрытьОкно()**, и присваивает значение **пусто** переданному объекту.

Эта функция обязательно должна быть вызвана для каждого графического окна, открытого программой, иначе при завершении программы возникнет исключение **ОшибкаВыполнения**.

Процедура грРазмерОкна()

процедура грРазмерОкна(вх окно: объект; вх горз, верт: целое);

Устанавливает размеры графического окна в точках.

Процедура `грНачать()`

```
процедура грНачать (вх окно: объект);
```

Начинает серию операций графического вывода.

Все последующие вызовы функций графического вывода не будут приводить к перерисовке окна до тех пор, пока серия не будет закончена вызовом `грЗакончить ()`. Таким образом можно заметно увеличить быстродействие программы и избежать ненужных мерцаний на экране.

Процедуры `грНачать ()` и `грЗакончить ()` поддерживают счётчик вложенности вызовов. Это означает, что серия будет закончена только тогда, когда процедура `грЗакончить ()` будет вызвана ровно столько раз, сколько до этого была вызвана процедура `грНачать ()`. Поэтому рекомендуется помещать вызов `грЗакончить ()` в секцию **напоследок** составного блока инструкций, например:

```
начало
  грНачать ();
начало
  грКруг(50, 50, 40);
  грПрямоугольник(0, 10, 40, 20);
  грОтрезок(0, 10, 40, 20);
напоследок
  // в этот момент все нарисованные фигуры
  // будут показаны на экране одновременно
  грЗакончить ();
конец;
окончание.
```

Процедура `грЗакончить()`

```
процедура грЗакончить (вх окно: объект);
```

Заканчивает серию операций графического вывода, начатую вызовом `грНачать ()` и выводит на экран все внесённые изменения.

Процедуры `грНачать ()` и `грЗакончить ()` поддерживают счётчик вложенности вызовов. Это означает, что серия будет закончена только тогда, когда процедура `грЗакончить ()` будет вызвана ровно столько раз, сколько до этого была вызвана процедура `грНачать ()`. Поэтому рекомендуется помещать вызов `грЗакончить ()` в секцию **напоследок** составного блока инструкций.

Процедура `грПеро()`

```
процедура грПеро (вх окно: объект; вх цвет: целое);
процедура грПеро (вх окно: объект; вх цвет, толщина: целое);
процедура грПеро (вх окно: объект; вх цвет, толщина, стиль: целое);
```

Устанавливает свойства пера — цвет, толщину и стиль. Свойства пера используются для прорисовки линий и контуров геометрических фигур.

Цвет должен быть указан в 24-битном формате RGB, где первый (младший) байт соответствует красному цвету, 2-й байт зелёному, а 3-й синему.

В параметре **стиль** можно передать одну из констант `грспXXXX`.

Процедура грКисть()

```
процедура грКисть (вх окно: объект; вх цвет: целое);  
процедура грКисть (вх окно: объект; вх цвет, стиль: целое);
```

Устанавливает свойства кисти — цвет и стиль. Свойства кисти используются для заливки геометрических фигур и в качестве фона при выводе текста.

Цвет должен быть указан в 24-битном формате RGB, где первый (младший) байт соответствует красному цвету, 2-й байт зелёному, а 3-й синему.

В параметре **стиль** можно передать одну из констант **грскXXXX**.

Процедура грШрифт()

```
процедура грШрифт (вх окно: объект; вх имя: строка);  
процедура грШрифт (вх окно: объект; вх имя: строка; вх размер: целое);  
процедура грШрифт (вх окно: объект; вх имя: строка; вх размер, стиль: целое);  
процедура грШрифт (вх окно: объект; вх имя: строка; вх размер, стиль, цвет: целое);
```

Устанавливает свойства шрифта — имя, размер, стиль и цвет. Свойства шрифта используются для вывода текста.

В параметре **имя** должно быть указано наименование шрифта. Если здесь передана пустая строка, то используется такой же шрифт, какой был выбран для вывода текста в отладочной консоли.

Параметр **размер** означает размер шрифта в пунктах. Если здесь передан нуль, то используется такой же размер шрифта, какой был выбран для вывода текста в отладочной консоли.

В параметре **стиль** можно передать нуль — это соответствует обычному тексту, — либо одно или несколько значений констант **грсшXXXX**, скомбинированных с помощью побитового ИЛИ. Например, выражение **грсшЖирный | грсшКурсив** означает жирный курсив, а выражение **грсшЗачерк | грсшПодчерк | грсшКурсив** означает зачёркнутый и подчёркнутый курсив.

Цвет должен быть указан в 24-битном формате RGB, где первый (младший) байт соответствует красному цвету, 2-й байт зелёному, а 3-й синему.

Функция грТочка()

```
функция грТочка (вх окно: объект; вх г, в: целое): целое;  
функция грТочка (вх окно: объект; вх г, в, цвет: целое): целое;
```

Возвращает цвет точки с указанными координатами в указанном окне. Если передан **цвет**, то для точки будет установлен новый цвет; при этом функция вернёт прежний цвет точки.

Цвет должен быть указан в 24-битном формате RGB, где первый (младший) байт соответствует красному цвету, 2-й байт зелёному, а 3-й синему.

Процедура грКруг()

```
процедура грКруг (вх окно: объект; вх г, в, радиус: целое);
```

Выводит в **окно** круг с указанным радиусом. В параметрах **г** и **в** должны быть переданы горизонтальная и вертикальная координаты центра круга.

Для прорисовки контура окружности используются установленные свойства пера, область внутри окружности заполняется в соответствии с установленными свойствами кисти.

Процедура грЭллипс()

```
процедура грЭллипс(вх окно: объект; вх г1, в1, г2, в2: целое);
```

Выводит в **окно** эллипс, вписанный в прямоугольник с указанными координатами. В параметрах **г1** и **в1** должны быть переданы горизонтальная и вертикальная координаты верхнего левого угла прямоугольника, а в параметрах **г2** и **в2** — правого нижнего.

Для прорисовки контура эллипса используются установленные свойства пера, область внутри эллипса заполняется в соответствии с установленными свойствами кисти.

Процедура грДуга()

```
процедура грДуга(  
    вх окно: объект; вх г1, в1, г2, в2: целое; вх начало, размер: дробное);
```

Выводит в **окно** дугу эллипса, вписанного в прямоугольник с указанными координатами. В параметрах **г1** и **в1** должны быть переданы горизонтальная и вертикальная координаты верхнего левого угла прямоугольника, а в параметрах **г2** и **в2** — правого нижнего.

Параметр **начало** определяет угол, с которого начинается дуга, а параметр **размер** — угловой размер дуги. Эти параметры должны быть указаны в радианах.

Для прорисовки дуги используются установленные свойства пера.

Процедура грСегмент()

```
процедура грСегмент(  
    вх окно: объект; вх г1, в1, г2, в2: целое; вх начало, размер: дробное);
```

Выводит в **окно** сегмент эллипса, вписанного в прямоугольник с указанными координатами. В параметрах **г1** и **в1** должны быть переданы горизонтальная и вертикальная координаты верхнего левого угла прямоугольника, а в параметрах **г2** и **в2** — правого нижнего.

Параметр **начало** определяет угол, с которого начинается сегмент, а параметр **размер** — угловой размер сегмента. Эти параметры должны быть указаны в радианах.

Для прорисовки контура сегмента используются установленные свойства пера, область внутри сегмента заполняется в соответствии с установленными свойствами кисти.

Процедура грСектор()

```
процедура грСектор(  
    вх окно: объект; вх г1, в1, г2, в2: целое; вх начало, размер: дробное);
```

Выводит в **окно** сектор эллипса, вписанного в прямоугольник с указанными координатами. В параметрах **г1** и **в1** должны быть переданы горизонтальная и вертикальная координаты верхнего левого угла прямоугольника, а в параметрах **г2** и **в2** — правого нижнего.

Параметр **начало** определяет угол, с которого начинается сектор, а параметр **размер** — угловой размер сектора. Эти параметры должны быть указаны в радианах.

Для прорисовки контура сектора используются установленные свойства пера, область внутри сектора заполняется в соответствии с установленными свойствами кисти.

Процедура `грОтрезок()`

```
процедура грОтрезок(вх окно: объект; вх г1, в1, г2, в2: целое);
```

Выводит в **окно** отрезок от точки с координатами **г1**, **в1** до точки с координатами **г2**, **в2**. Для прорисовки отрезка используются установленные свойства пера.

Процедура `грЛоманая()`

```
процедура грЛоманая(вх окно: объект; вх точки: МассивТочек);
```

Выводит в **окно** ломаную линию. В параметре **точки** должны быть переданы пары координат для каждой вершины ломаной линии. Для прорисовки звеньев ломаной используются установленные свойства пера.

Процедура `грПрямоугольник()`

```
процедура грПрямоугольник(вх окно: объект; вх г1, в1, г2, в2: целое);  
процедура грПрямоугольник(вх окно: объект; вх г1, в1, г2, в2, р: целое);  
процедура грПрямоугольник(вх окно: объект; вх г1, в1, г2, в2, рг, рв: целое);
```

Выводит в **окно** прямоугольник. Параметры **г1**, **в1** определяют координаты верхнего левого угла прямоугольника, а параметры **г2**, **в2** — правого нижнего.

В параметре **р** можно передать радиус закругления углов прямоугольника. Если переданы **рг** и **рв**, то **рг** определяет радиус закругления по горизонтали, а **рв** — по вертикали.

Для прорисовки контура прямоугольника используются установленные свойства пера, область внутри прямоугольника заполняется в соответствии с установленными свойствами кисти.

Процедура `грМногоугольник()`

```
процедура грМногоугольник(вх окно: объект; вх точки: МассивТочек);
```

Выводит в **окно** многоугольник. В параметре **точки** должны быть переданы пары координат для каждой вершины многоугольника. Для прорисовки сторон многоугольника используются установленные свойства пера, область внутри многоугольника заполняется в соответствии с установленными свойствами кисти.

Функция `грТекст()`

```
процедура грТекст(вх окно: объект; вх г, в: целое; вх текст: строка);
```

Выводит в **окно** переданный **текст**. Координаты верхнего левого угла прямоугольника с текстом должны быть переданы в параметрах **в**, **г**. Для вывода текста используются свойства шрифта, установленные вызовом `грШрифт()`, фон прямоугольника заполняется в соответствии с текущими свойствами кисти.

Разное

Процедура уничтожить()

процедура уничтожить (vv o: объект);

Уничтожает переданный объект, выполняя все необходимые действия, связанные с его уничтожением, и присваивает переданному параметру значение **пусто**.

Если переданный объект не существует — т. е., он был ранее уничтожен, никогда не был создан, либо в параметре было передано значение **пусто**, — процедура будет выполнена без ошибок.

Процедура пауза()

процедура пауза (vx msec: целое);

Приостанавливает выполнение программы на переданное число миллисекунд.